

Michael Orkin Ed Bogas

SURVIVAL on PLANET X

with the ATARI® Home Computer



SURVIVAL ON PLANET X

SURVIVAL ON PLANET X

WITH THE ATARI® HOME COMPUTER

Michael Orkin
Ed Bogas

Illustrated by William Luckey



A Reston Computer Group Book
Reston Publishing Company, Inc.
A Prentice-Hall Company
Reston, Virginia

Library of Congress Cataloging in Publication Data

Orkin, Michael.

Survival on Planet X with the ATARI® Home Computer.

(A Reston computer group book)

Includes index.

1. Atari 400 (Computer) — Programming. 2. Atari 800 (Computer) — Programming. 3. Atari 1200XL (Computer) — Programming. 4. BASIC (Computer program language)

I. Bogas, Ed. II. Title.

QA76.8.A8074 1984 001.64'2 83-16056

ISBN 0-8359-7412-X

Copyright© 1984 by Reston Publishing Company, Inc.,
A Prentice-Hall Company,
Reston, Virginia 22090

All rights reserved. No part of this book may be reproduced in any way, or by any means, without written permission from the publisher.

The final typeset pages were produced on a TyXSET 1000 system in Reston, Virginia, using a Mergenthaler Omnitech/2100. The galley and page proofs were produced on the TyXSET 1000 system using a Canon LBP-10 Laser Printer.

TyXSET 1000 is a trademark of TyX Corporation
ATARI is a registered trademark of Atari, Inc.

10 9 8 7 6 5 4 3 2 1

Printed in the United States of America

CONTENTS

Preface	vii
Chapter 1: A CALL FOR HELP	1
Chapter 2: XANG	9
Chapter 3: THE LEADERSHIP COUNCIL	21
Chapter 4: ESCAPE	31
Chapter 5: FLIGHT TO THE HILLS	49
Chapter 6: XAVIAR	61
Chapter 7: THE TROLLEY	73
Chapter 8: THE ARMY LIZARDS	85
Chapter 9: BACK TO THE SPACESHIP	95
Chapter 10: ANOTHER CALL FOR HELP	111
Glossary of Programming Terms	125
Index of Programming Terms	151

THE HISTORY OF THE
CITY OF BOSTON

1	1630	1630
2	1630	1630
3	1630	1630
4	1630	1630
5	1630	1630
6	1630	1630
7	1630	1630
8	1630	1630
9	1630	1630
10	1630	1630
11	1630	1630
12	1630	1630
13	1630	1630
14	1630	1630
15	1630	1630
16	1630	1630
17	1630	1630
18	1630	1630
19	1630	1630
20	1630	1630
21	1630	1630
22	1630	1630
23	1630	1630
24	1630	1630
25	1630	1630
26	1630	1630
27	1630	1630
28	1630	1630
29	1630	1630
30	1630	1630
31	1630	1630
32	1630	1630
33	1630	1630
34	1630	1630
35	1630	1630
36	1630	1630
37	1630	1630
38	1630	1630
39	1630	1630
40	1630	1630
41	1630	1630
42	1630	1630
43	1630	1630
44	1630	1630
45	1630	1630
46	1630	1630
47	1630	1630
48	1630	1630
49	1630	1630
50	1630	1630
51	1630	1630
52	1630	1630
53	1630	1630
54	1630	1630
55	1630	1630
56	1630	1630
57	1630	1630
58	1630	1630
59	1630	1630
60	1630	1630
61	1630	1630
62	1630	1630
63	1630	1630
64	1630	1630
65	1630	1630
66	1630	1630
67	1630	1630
68	1630	1630
69	1630	1630
70	1630	1630
71	1630	1630
72	1630	1630
73	1630	1630
74	1630	1630
75	1630	1630
76	1630	1630
77	1630	1630
78	1630	1630
79	1630	1630
80	1630	1630
81	1630	1630
82	1630	1630
83	1630	1630
84	1630	1630
85	1630	1630
86	1630	1630
87	1630	1630
88	1630	1630
89	1630	1630
90	1630	1630
91	1630	1630
92	1630	1630
93	1630	1630
94	1630	1630
95	1630	1630
96	1630	1630
97	1630	1630
98	1630	1630
99	1630	1630
100	1630	1630

PREFACE

Survival on Planet X is an adventure story that includes computer programs illustrating events in the story. As designers of video games, we have tried to present the basic ideas of programming in an exciting way. The programs are written in the BASIC programming language and can be used on an ATARI® home computer. You don't have to know anything about computers to use the programs. In fact, you can read the story without using a computer at all.

Some of the programs in this book are easy to understand and some are complicated. All are short enough to easily type into your computer. In the back of the book is a glossary of programming procedures that you can use as a guide.

Computers are not as smart as people and can only understand precise instructions. When you use the programs in the book, you must be sure to type them into the computer exactly as they are printed. Even a missing comma or misplaced quotation mark can hopelessly confuse the computer.


Programs used for video games, business, or scientific applications may be longer and more complicated than the programs in the book, but the ideas are the same. We hope you enjoy *Survival on Planet X* and have fun experimenting with your computer.



A CALL FOR HELP

Not wanting to spend another dull summer babysitting and visiting relatives, Vivian decided to take her spaceship on a trip through an uncharted solar system. She enjoyed being alone and was able to relax in the silence of outer space.

One afternoon (at least it was afternoon somewhere on Earth), just after Vivian put her ship on automatic control in order to watch a meteor storm, she noticed the message "HELP!" on her computer communications screen. She immediately used her computer to trace the message and discovered that it came from a small, unexplored planet named Planet X. Vivian decided to investigate and changed course. Soon she was heading toward the small planet.



We will use the ATARI home computer to help describe Vivian's adventure. As a result, you will learn about computer programming. The programming language we will use is called ATARI BASIC. Load your BASIC cartridge and turn on your computer. "READY" will appear in the upper left-hand corner. You can get the computer to print "HELP" by typing

```
PRINT "HELP"
```

and pressing the **RETURN** key. Try it.

To get the computer to repeatedly print "HELP" the way it appeared on Vivian's computer, we need a program. A *computer program* is a set of instructions that tells the computer what to do.

Here are some guidelines you can refer to throughout the story. Whenever you finish typing a line in a program, you must press the **RETURN** key before you start the next line. To correct a mistake before you are finished typing a line, use the **DELETE BACK S** key which backspaces and erases characters. (Find this key on your keyboard.) To correct an already typed line, you can retype it. The computer will automatically put it in its proper place. If a program has a mistake in it that you can't find, you can erase the program from the com-

puter's memory and start over. You do this by typing **NEW** and pressing the **RETURN** key or by turning the computer off and turning it on again. To clear the screen without erasing the computer's memory, hold down the **SHIFT** key and press the **CLEAR** key.

Now type the following program. Press **RETURN** after each line.

```
10 PRINT "HELP! "  
20 GOTO 10
```

This program puts the computer in an unending "loop." Once it starts, you can stop it by pressing the **BREAK** key or the **SYSTEM RESET** key. To start the program, type **RUN** and press **RETURN**. Do this now.

Line 20 in the above program instructs the computer to go back to line 10, where it started. Therefore, the computer keeps printing **HELP** until you stop it. Type **LIST** and press **RETURN**. The computer will print a list of the program. Now change line 10 by adding a semicolon after the quotation mark. Type

```
10 PRINT "HELP! " ;
```

RUN this program. The semicolon at the end of a **PRINT** statement tells the computer to carry out the next

PRINT statement at the next available space. If there is no punctuation mark, as before, the computer skips to the next line before printing again.

Now change line 10 once more by typing a comma instead of a semicolon at the end of the PRINT statement. Type

```
10 PRINT "HELP! ",
```

Now RUN the program.

Finally, we will add a buzzing sound to the program. You can make the computer buzz by using the **CTRL** ("Control") key located on the left side of your keyboard, along with the number 2. Press the **CTRL** key and, while it is still pressed, type the number 2.

To use the buzzer in a program, we need the **ESC** key, located on the upper left-hand side of the keyboard. We will have the buzzer sound each time HELP! is printed. Retype line 10 as follows: After typing HELP!, press the **ESC** key, then press the **CTRL** key and, while the **CTRL** key is still pressed, type the number 2. The symbol "↑" should appear on the screen when you do this. Type line 10 as follows and RUN the revised program.

```
10 PRINT "HELP! ↑ ",
```

After two long days of travel, Vivian reached Planet X. She landed her spaceship in a valley surrounded by large, rocky mountains. Vivian's computer informed her that she could breathe the air. Emerging from her spaceship, she met a strange creature who looked something like a lizard but stood upright. Vivian was a bit scared but reassured herself that she could at least understand the lizard creature with the help of the translating system on her computer.

"My name is Xang," said the creature who greeted Vivian. "My job is to welcome visitors to Planet X. Welcome to Planet X! There are very few visitors to Planet X. In fact, I've never welcomed anyone before. I usually sit in my office all day and wait for someone to arrive. You can't imagine how excited I am."

Type NEW to erase the current program from the computer's memory. We will write a new program that shows Xang's excitement. Enter the following program:

```
10 PRINT "A VISITOR! "  
20 PRINT  
30 PRINT "I CAN'T BELIEVE IT! "  
40 PRINT  
50 GOTO 10
```



RUN this program. The PRINT statement alone tells the computer to skip a line. Now type the following lines. They cause a delay in Xang's speech. We will call the statements *delay statements*. The computer will automatically put them in the right sequence.

```
25 FOR N = 1 TO 500: NEXT N
45 FOR N = 1 TO 500: NEXT N
```

RUN this program. Xang seems to have calmed down a little. Now LIST the program. Notice that the statements are in the right sequence. It is a good idea to number statements at least by 10's so you will have room to insert new statements.

Vivian had never seen a lizard creature before and was happy to see that he was friendly. She wondered who had sent the HELP message.

XANG

After Xang welcomed Vivian to Planet X, he handed her a form and said, "You must fill out a Planet X visitor's card which all visitors to Planet X are required to complete."

Here are some programs that show what a Planet X visitor's card looks like. The first program prints a blank form with a heading. To clear the computer's memory, type **NEW** and press **RETURN**. Then type the following program and run it:

```
10 PRINT "PLANET X VISITOR ' S CARD"  
20 FOR N = 1 TO 18  
30 PRINT  
40 NEXT N
```

Lines 20 through 40 in the above program are called a *FOR/NEXT loop*. After the computer reads line 20, it performs every instruction until the **NEXT** statement and then returns to line 20.

Every time it does this, the computer increases the value of **N** by 1 until **N** equals 18. In this way, the letter **N** acts as a "counting variable." You can use any letter for the counting variable. When the loop has been executed



18 times, the computer moves on to the next instruction if there is one. In this program, the only instruction in the FOR/NEXT loop is the PRINT statement in line 30. Since a blank PRINT statement causes the computer to skip a line, the loop causes the computer to skip 18 lines. In the last chapter, we used a FOR/NEXT loop with no instructions between the FOR and NEXT statements:

```
FOR N = 1 TO 500: NEXT N
```

The computer went through the loop 500 times without doing anything. The time it took the computer to do this caused the delay between Xang's exclamations. During each brief pause, the computer was actually reading the FOR/NEXT instructions 500 times. In ATARI® BASIC, you can put more than one instruction on a single line if you separate each instruction by a colon. We did this for the delay statement.

Now we will change the Visitor's Card. Type the following lines and RUN the program:

```
12 PRINT  
14 PRINT "VISITOR#", "NAME", "HOME"  
30 PRINT N, ". . . . .", ". . . . ."
```

When a comma occurs in a PRINT statement, it skips

to the next position, like a tab key on a typewriter. In ATARI® BASIC, the tab settings are every 10 spaces across a line. The comma in the PRINT statement makes it easy to write a program that prints information in columns. In the last chapter, the comma was used in a PRINT statement to make the HELP message have an interesting pattern. This Visitor's Card numbers the lines that visitors are supposed to sign and has dotted lines to show where they are supposed to sign it. LIST the program. It should look like this:

```
10 PRINT "PLANET X VISITOR ' S CARD"  
12 PRINT  
14 PRINT "VISITOR# ", "NAME", "HOME"  
20 FOR N = 1 TO 18  
30 PRINT N, ". . . . .", ". . . . ."  
40 NEXT N
```

The Visitor's Card Xang handed Vivian was blank. "You might be the first one who has ever visited Planet X," he said. "Here is a demonstration card you can look at in case you need help filling out the card."

To see the demonstration card, change line 30 as follows and RUN the revised program.

```
30 PRINT N, "XANG", "PLANET X"
```

Vivian completed the Visitor's Card and Xang asked her to pronounce her name. He told her, "I know a poem about someone named Vivian" and recited a poem with her name in it.

Information such as your name is stored in the computer's memory the way you would put something in a box and label the box so you know where to find it. This storage location with a label is called a *variable*. If you are storing nonnumeric information, the variable is called a *string variable*. A string variable can be denoted by a letter followed by a dollar sign, like A\$ or B\$. You can also denote a string variable with a word, like NAME\$ or ADDRESS\$. When you use a string variable in a program, you must also use a DIM statement (DIM stands for "dimension") which gives the computer

a maximum length for the variable. For example, DIM A\$(100) tells the computer that the string variable A\$ will be no more than 100 characters long. (A character is a letter, number, or other symbol.)

One method of putting information into the computer's memory is by using an INPUT statement. When the computer reads an INPUT statement, it will print a question mark on the screen and wait until you type in the information you want to input. When you do this and press **RETURN**, the computer will proceed to the next instruction.

Here is a program that uses the INPUT statement. Type NEW before entering the program so the current program will be erased. Then enter the program and RUN it.

```
10 DIM A$ (100)
20 PRINT "WHAT IS YOUR NAME? "
30 PRINT
40 INPUT A$
50 PRINT A$; "? THAT'S FUNNY. "
60 PRINT "I KNOW A POEM ABOUT"
70 PRINT "SOMEONE NAMED "; A$
```

Notice that in lines 50 and 70, A\$ is not included in

quotes. If it were, the computer would print "A\$" rather than the current value of the variable A\$, namely, your name.

Now we will instruct the computer to print the poem with your name. First, we will use a FOR/NEXT loop to skip some lines. Add the following lines and RUN the revised program.

```
80 FOR N = 1 TO 7
90 PRINT
100 NEXT N
110 PRINT "THERE WAS ONCE A RULER"
120 PRINT "NAMED "; A$; " THE THIRD"
130 PRINT "WHO RULED A LARGE KINGDOM"
140 PRINT "WITHOUT SAYING A WORD"
```

If you have trouble getting the program to run, LIST it and check to see that it is correctly typed. Even a forgotten or incorrect punctuation mark could keep a program from running. Now complete the poem by adding the following lines. Then RUN the program.

```
150 PRINT "WHEN THE PEOPLE CRIED "; A$
160 PRINT "LET YOUR RULINGS BE HEARD"
170 PRINT "OFF WITH YOUR HEADS! "
180 PRINT "SAID "; A$; " THE THIRD. "
```


LIST the program to see it all at once. RUN the program a few times, inputting a different name each time.

Vivian was pleased to hear a poem about herself and told Xang, "I really enjoyed the poem, but I'm here to investigate a signal for help that I received in space." She asked him if he knew anything about it. Xang replied that his job was to greet visitors to Planet X and that he didn't really know much about anything else.

"The Leadership Council would know about things like signals for help," said Xang. "I will take you to see them."

He walked over to a vehicle that looked like a small car with a propeller on the back. He climbed into the car and motioned for Vivian to sit next to him. She did this with some difficulty, since the vehicle was built for lizard creatures, not Earthlings. Xang started the propeller car, which made a loud whirring sound, and he and Vivian sped off down a bumpy road.

As she and Xang traveled, Vivian was able to see the countryside of Planet X. A bright sun shone in the Planet X sky. Tall, greenish-purple trees that looked like painted weeping willows were scattered along the dirt road. A stream with purple water flowed along the roadside. A pair of strange animals that resembled giant hamsters walked slowly along the bank of the






stream. Xang told Vivian that these animals were called xamsters. Thin green birds with huge eyes and long beaks flew by overhead. A small blue animal like a cat darted behind a tree. Some lizard creatures who looked like Xang were walking along the side of the road carrying what appeared to be garden tools.

One of the creatures was holding a bushy purple plant. As Vivian and Xang drove by, the creatures stared straight ahead with blank expressions.



Further on, the travelers passed another propeller car. There were about six lizard creatures in it. Its propeller wasn't spinning; it was being pulled by a xamster.

Abandoned in the middle of a field, Vivian saw a piece of farm machinery that looked like a tractor. The tractor had not been used for some time and was missing all of its wheels. Plants were growing from it and it was covered with purple rust. A few lizard creatures were sitting around a small fire next to the tractor. They appeared to be cooking something. In the distance there



was a rundown house. A propeller car with the front end smashed in was in a dirt driveway next to the house.

After about an hour's travel, Vivian and Xang came to a city located at the base of a mountain. A market in the center of town was crowded with lizard creatures carrying baskets and pulling carts. Larger carts and propeller cars were being pulled through the streets by xamsters. Other strange creatures wandered through the streets among the lizard creatures. Finally, Xang parked the propeller car next to a large stone building that seemed on the verge of collapse.

"This is government headquarters," said Xang proudly. "The Leadership Council is here."

THE LEADERSHIP COUNCIL

Vivian and Xang entered government headquarters and were met at the door by an armed lizard guard. Xang explained that Vivian was a visitor to Planet X and that she wanted to see the Leadership Council. The guard rudely informed Xang that no visitor to Planet X had ever been allowed to see the Leadership Council. Vivian pointed out that the reason no visitors had ever been allowed to see the Leadership Council was that there had never been any visitors to Planet X before.

The guard looked at Vivian suspiciously and told her that he did not have the authority to grant her request and that, if she wanted to discuss the matter further, she would have to speak with Head Administrator #1 whose office was down the hall and to the right. "I guess that's what we'll have to do," said Vivian, who turned and walked down the hall with Xang following close behind.

Vivian and Xang entered the office of Head Administrator #1. Xang greeted the administrator and said, "I have brought a visitor who requests to see the Leadership Council."

"No visitor has ever been allowed to see the Leadership Council," said Head Administrator #1. "I don't have the authority to grant such a request. Try Head Administrator #2. His office is down the hall and to the right."

"This is worse than school," said Vivian, who hated it when people couldn't make decisions. With Xang close behind, she

marched down the hall and to the right to the office of Head Administrator #2. Unfortunately, Head Administrator #2 gave the same response as Head Administrator #1. Before long, Vivian and Xang found they were in the office of Head Administrator #3. After getting the same response there, Vivian and Xang went to the office of Head Administrator #4, who proceeded to send them down the hall and to the right and back to the office of Head Administrator #1. After walking around the halls until they were quite dizzy, Vivian and Xang reached the office of Head Administrator #4 for the sixth time.

Type and RUN the following program;it will demonstrate Vivian's difficulty in trying to find an administrator who would grant her request to see the Leadership Council. Clear the computer's memory before starting. If a line in the program is longer than a line on the screen, the computer will automatically go to the next line. We will explain the SOUND statements later.

```
20 FOR X = 1 TO 4
30 PRINT "HI, I'M ADMINISTRATOR # "; X
40 PRINT
```

```
50 FOR L = 1 TO 200: NEXT L
70 FOR W = 1 TO 25
80 PRINT "WALK" ,
90 SOUND 0, 0, 10, 15
100 FOR L = 1 TO 50: NEXT L
120 SOUND 0, 115, 3, 4
130 FOR L = 1 TO 50: NEXT L
150 NEXT W
160 PRINT
170 PRINT
180 NEXT X
```

List the program and try to understand its logic. Notice that lines 50, 100, and 130 are delay statements. To visit each of the four administrators six times, add these lines and RUN the program.

```
10 FOR K = 1 TO 6
190 NEXT K
```

To experiment with different sounds, you can vary the second number in the SOUND statement of line 90. This number can range from 0 to 255. For example, try SOUND 0,80,10,4 for line 90 and RUN the revised program.



When Vivian and Xang reached the office of Head Administrator #4 for the sixth time, she told him that although it was true that no visitor to Planet X had ever been given permission to see the Leadership Council, it was also true that no visitor had ever been refused permission to see the Leadership Council. If the administrator refused to grant her permission, he would be the first one to ever refuse to allow a visitor to see the Leadership Council.

After thinking about this, Head Administrator #4 said, "I think you may be correct and I grant your request. The Leadership Council is upstairs and down the hall."

Vivian and Xang went upstairs and down the hall and entered the office of the Leadership Council. A lizard creature with long lizard hair and big lizard eyes greeted them. "My name is Xanada," said the lizard creature. "It is my job to make sure that everyone who sees the Leadership Council is intelligent. Therefore, I will give you a test. There are five questions. If you give a wrong answer you must go to jail. First question: How much is 36,287 plus 48,898?"

"Wait a minute," said Vivian. "I only came here to answer a signal for help. I don't want to go to jail. I think I'll just leave now." Vivian turned to go, but was stopped by two armed lizard guards. Xang was nowhere to be seen.

"I'm sorry," said Xanada, "but you can't leave now."

Realizing that she had no choice, Vivian used her pocket computer to quickly solve the first problem.

In BASIC you can do arithmetic by using the PRINT statement. The symbol for addition is "+"; for subtraction it is "-"; for multiplication it is "*"; and for division it is "/". For example, to add 8 plus 9, just type

```
PRINT 8 + 9
```

and press **RETURN**. To multiply 17 times 12, just type

```
PRINT 17 * 12
```

and press **RETURN**. Experiment with the PRINT statement to do more arithmetic problems.

We will write a program that instructs the computer to add any two numbers. Just as string variables are used to input words, *numeric variables* are used to input numbers. They can be denoted by letters like X or Y. You don't need a DIM statement with numeric variables. Clear the computer's memory. Enter the following program and RUN it several times, each time adding two different numbers. The GOTO statement allows you to keep using the program until you press the **BREAK** key.

```
10 PRINT "THIS PROGRAM ADDS TWO NUMBERS"
```

```
20 PRINT "INPUT FIRST NUMBER"
```

```
30 INPUT X
```

```
40 PRINT "INPUT SECOND NUMBER"
```

```
50 INPUT Y
60 PRINT
70 PRINT "THE ANSWER IS "; X + Y
80 PRINT
90 GOTO 20
```

Vivian correctly answered the first problem and Xanada gave her the second problem: "How much is 87,456 minus 63,084?"

The addition program can be easily changed to a subtraction program. Make the following changes and RUN the subtraction program. Use it to answer Xanada's second question.

```
10 PRINT "THIS PROGRAM SUBTRACTS THE SECOND NUMBER FROM THE FIRST"
70 PRINT "THE ANSWER IS "; X - Y
```

After Vivian answered the second problem correctly, Xanada gave her the third problem: "How much is 2,867 times 1,923?"

Now we will change the subtraction program to a multiplication program. Sometimes, when numbers are too large or too small to print in the usual way, the computer prints them in a peculiar looking form called *exponential notation*. To convert a number in exponential notation to a number in ordinary form, you move the decimal point over as many spaces as are indicated after the E: For example, 2.31E4 indicates that the decimal point should be moved four places to the right: 23,100. The notation 81.22E-5 means to move the decimal point five places to the left: .0008122. Make the following changes and RUN the resulting program. Use it to answer Xanada's third question.

```
10 PRINT "THIS PROGRAM MULTIPLIES TWO NUMBERS"  
70 PRINT "THE ANSWER IS "; X * Y
```

List the program and experiment with it until you understand how it works.

Vivian answered the third problem and Xanada gave her the fourth one: "How much is 619,203 divided by 39?"

Make the indicated changes and RUN this program to answer Xanada's fourth question.

```
10 PRINT "THIS PROGRAM DIVIDES THE FIRST NUMBER  
BY THE SECOND"  
70 PRINT "THE ANSWER IS "; X/Y
```

When Vivian answered the fourth question correctly Xanada said, "Here is the fifth question: Who was the first ruler of Planet X?"

"How can I answer that?" exclaimed Vivian. "I never studied the history of Planet X!"

"Enough of this insolence!" snapped Xanada. She turned to the lizard guards and said, "Take the visitor to jail!"

Before Vivian could reply, the two burly lizard guards grabbed her arms and took her away. The guards led her down two flights of stairs to the basement of the building. The jail was located at the end of a dimly lit hall behind a purple steel door. The guards tossed Vivian into a dark, cramped cell that was built for lizard creatures, not humans. As she lay in the dark, Vivian thought that maybe babysitting and visiting relatives wasn't such a bad way to spend the summer after all.



ESCAPE

When her eyes got used to the light, Vivian saw that the only furniture in her cell was a small cot. There were two other lizard creatures in the cell next to hers. One was thin and the other was plump. They seemed to be arguing. As Vivian moved toward them, she discovered that they were playing a game. She heard this conversation:

Thin Lizard: Ready—one, two, three...

Both Lizards (together): AIR!

Plump Lizard (angrily): 3,974,425 to 72

Thin Lizard: Ready—one, two, three...

Both Lizards: LAND!

Plump Lizard: 3,974,426 to 72

Thin Lizard: Ready—one, two, three...

Both Lizards: AIR!

Plump Lizard: 3,974,427 to 72

Thin Lizard: Ready—one, two, three...

Both Lizards: LAND!

Plump Lizard: I quit! No matter what strategy I use, you always know what I'm going to say.

Thin Lizard: It's pointless to quit, Xax, since there's nothing else to do. Try a different strategy.



Plump Lizard: All right. But if you keep winning I'll quit even if it means we have to stare at the wall all day.

Vivian interrupted the lizard creatures and asked them what game they were playing.

"It's a war game," said the thin lizard creature, whose name was Xon. "The defender tries to guess how the invader will attack. I'm the defender and Xax is the invader. At the count of three, we each can say 'AIR' or 'LAND.' If I say the same thing as Xax, then I have guessed his attack strategy and I win. If we say different things — that is, if one of us says 'AIR' and the other says 'LAND' — he has avoided my defense and he wins. Right now I'm ahead by a score of 3,974,428 to 72."

Vivian asked Xax what kind of strategy he used in deciding how to attack.

"Well, first I was attacking by AIR all the time," said Xax, "but Xon got wise to that, so I started attacking by LAND all the time. He caught on to that too, so I started alternating and saying AIR-LAND-AIR-LAND but he figured that out as well. Now I don't know what to do."

"You could use a random strategy," Vivian told Xax. "That way it would be impossible to predict for sure whether you would invade by air or land. In the long run, Xon could only be right about half the time." Vivian then used her computer to demonstrate a random strategy.

We will write a program for the AIR-LAND invasion game in which the computer is the invader and you are the defender. Remember the rules: If you correctly guess the invader's route (AIR or LAND), you win. If your guess is incorrect, you lose. The computer will pick an invasion route at random so you will not be able to predict whether it will choose AIR or LAND.

The BASIC function RND(0) generates a "random" decimal fraction. A sequence of numbers generated this way has no predictable pattern. To see a sequence of these random numbers, enter and RUN this program. (If the number is very small, the computer will use exponential notation.)

```
10 PRINT RND(0)
20 FOR J = 1 TO 100: NEXT J
30 GOTO 10
```

You can eliminate the fractional part of a number by using the INT function. INT(X) drops the fractional part of X. For example, INT(3.1416) = 3 and INT(.3679) = 0.

If you multiply RND(0) by 2 you will get a random number between 0 and 2. You can change the random number 2*RND(0) into a "random digit," 0 or 1, by using the INT function to drop its fractional part. There is a

50-50 chance that this random digit will be 0 and a 50-50 chance that it will be 1. This is just like tossing a coin, in which there is a 50-50 chance that heads will come up and a 50-50 chance that tails will come up. To see some random digits, change line 10 as follows and RUN the revised program. The semicolon is included at the end of the PRINT statement so that the random digits will be printed across the line instead of in a column.

```
10 PRINT INT ( 2 * RND ( 0 ) ) ;
```

We will instruct the computer to invade by AIR if the random digit $\text{INT}(2 * \text{RND}(0))$ equals 0 and to invade by LAND if $\text{INT}(2 * \text{RND}(0))$ equals 1. This is like tossing a coin and picking AIR if heads comes up and LAND if tails comes up. To do this, we will use the IF-THEN statement, which instructs the computer to follow an instruction if a specified condition is met.

In order to have a variable equal a specified value, we can use the LET statement. For example, if we wanted to have $X = 20$ we could use the instruction

```
LET X = 20
```

In fact, it is not even necessary to use the word LET. We could simply type

X = 20

and the computer would understand. In the next program we will let X equal a random number.

Clear the computer's memory, then enter and RUN the following program, in which the computer will print AIR if INT(2*RND(0)) equals 0 and will print LAND if INT(2*RND(0)) equals 1.

```
10 X = INT (2*RND (0) )
20 IF X = 0 THEN PRINT "AIR"
30 IF X = 1 THEN PRINT "LAND"
40 FOR J = 1 TO 200: NEXT J
60 GOTO 10
```

The computer now picks an invasion route at random. It can't tell you this in advance or you will know how to defend yourself, so it will keep the information secret, stored in its memory as the variable B\$. Enter these lines:

```
10 PRINT "AIR-LAND INVASION"
15 PRINT
20 DIM A$(100) , B$(100)
25 X = INT (2*RND (0) )
```

```
30 IF X = 0 THEN B$ = "AIR"  
40 IF X = 1 THEN B$ = "LAND"
```

After the computer chooses an invasion route, you must choose a defense. Add the following lines to the program. Line 65 ensures that you correctly enter AIR or LAND as your defense before the computer proceeds.

```
50 PRINT "INPUT DEFENSE ( 'AIR' OR 'LAND' ) "  
60 INPUT A$  
65 IF A$<>"AIR" THEN IF A$<>"LAND" THEN GOTO 50  
70 PRINT
```

After this, the computer must tell you its invasion route and who won. The inequality symbols "<>" together mean "not equal to." Add these lines and play the game:

```
80 PRINT "I CHOOSE TO INVADE BY "; B$  
90 PRINT  
100 IF B$ = A$ THEN PRINT "YOU WIN"  
110 IF B$ <> A$ THEN PRINT "I WIN!! "  
120 PRINT  
130 GOTO 25
```

Vivian suggested to Xax that on each play he could toss a coin to decide whether to attack by AIR or LAND. Xax thanked Vivian for her help. Then, he looked at her curiously and said, "By the way, who are you? You don't look like a lizard."

Vivian told Xax and Xon who she was and why she came to Planet X. When Xon heard about the help message he said, "You're the visitor from the sky ! You're the one who Xaviar promised would be sent to save us!"

Vivian asked Xon who Xaviar was and what she was going to save them from. Xon replied, "Our civilization was once a prosperous one. Machines did our work for us. Propeller cars carried us wherever we wanted to go and farm machines harvested our crops. We had machines that cooked and machines to build things. Our lives depended on machines. Then, we ran out of fuel. The Leadership Council didn't tell anyone about the shortage until it was too late. They hoarded what remained for themselves. They thought they would eventually find more fuel but they didn't."

"Now animals pull our cars and the machines are useless. The Leadership Council puts anyone who disagrees with them in jail. I'm here because I told people we should be looking for new kinds of fuel. Xax is in jail because he refused to work. He used to be a mechanic and now there are no jobs for mechanics. We belong to the Rebel Forces. Xaviar is our leader."

"We live in the mountains in caves. Xaviar knows magic and all of his wishes come true. He has told us that a visitor will

come from the sky to save us. You are the visitor. Long live Xaviar! Long live Xaviar!"

Vivian was not sure how she was supposed to help Xaviar and his followers or, for that matter, anyone else on Planet X. She decided she had better try to escape from jail and get back to her spaceship. She went to the cell door and examined the lock. It was a combination lock marked with the Planet X numbers from 1 to 30. It worked like an Earth combination lock and would open after turning the lock to three numbers. Since there were 30 possible numbers for each rotation, there were $30 \times 30 \times 30$ or 27,000 possible number combinations, only one of which was correct. If it took 20 seconds to try a combination, it would take 150 hours or about one month working five hours a day to try all 27,000 combinations. Vivian knew that she couldn't spend that much time trying to open the lock. Fortunately, her pocket computer had a vibration detector so that it could detect when a correct number put the lock mechanism in alignment. Vivian attached the computer to the lock and within minutes she found the right combination. With a big smile, she opened the cell door.

We will write a program, using FOR/NEXT loops, that demonstrates the speed of the computer in trying all the 27,000 possible combinations. To start, we will make the second and third numbers we try equal to 1 and let

the first number range from 1 to 30. Enter and RUN this program:

```
10 FOR J = 1 TO 30
20 PRINT J, 1, 1
30 NEXT J
```

Now we will let the second number range from 1 to 30 as well. Make the following changes and RUN this program:

```
15 FOR K = 1 TO 30
20 PRINT J, K, 1
25 NEXT K
```

Finally, we will let the third number range from 1 to 30. Make the following changes and RUN the finished program.

```
17 FOR L = 1 TO 30
20 PRINT J, K, L
23 NEXT L
```

If you press the **BREAK** key to stop a program, you can restart it by typing **CONT** and pressing **RETURN**. Try starting and stopping the lock combination program as it is being run.

When Vivian opened the lock to her cell, Xax called out, "Hey, what about us? Open our cell too." Vivian wanted to get back to her spaceship as quickly as possible, but she couldn't let Xax and Xon stay in jail. She attached her computer to the lock on Xax and Xon's cell and soon they were free. The three tiptoed down the dark hall and, when they were sure no one was looking, they dashed up the stairs and out a back door.



When she got outside the building, Vivian ran down the street with Xax and Xon close behind. As they rounded the corner, a group of lizard guards came out of the building.

"STOP!" shouted a guard as he and the other guards ran toward the street.

"WAIT!" shouted Xax and Xon as they ran after Vivian.

Vivian ran through a narrow alley and came out on a street lined with a series of small shops. She ran into a shop with a sign on the front that said "XELDA'S TOURS." Moments later Xax and Xon ran by the shop, not noticing that Vivian had gone inside. Soon after, the guards went running by.

Vivian found herself in a room in which the walls were filled with pictures and maps. There was a picture of a lizard woman lying on a rock next to a swimming pool. Another picture showed a lizard family standing in front of a large hotel. There was another picture of what looked like a lizard amusement park.

At the far end of the room was a large table covered with more maps and pictures. Behind the table was a desk. Sitting at the desk was a creature who looked like an overweight hen. She was smoking a long green cigarette. On her desk was an ashtray filled with green cigarette butts. Her face was covered with multicolored makeup and she was wearing purple beakstick. The hen creature was reading a magazine called "Travel." There was a nameplate on her desk that said "XELDA."

"What can I do for you, honey?" said Xelda when she saw Vivian.

"Hi," said Vivian, looking nervously at the front door. "I'm wondering if you have tours to uhh... well, actually, I left my spaceship not far from here and I need a ride. And, if you can't do that, do you have any tours to Earth? You see, I need to get home in a hurry."

"I don't know where Earth is, but the xamster express leaves for the other side of the mountain every Monday morning," said Xelda as she put out her cigarette in the ashtray and lit another one. "We also have a scenic 10-day cruise for 150 xents."



Get away from the pressures of the city. There's dancing on deck every night to the live music of Xandy and the Leaping Lizards. Every day there's a new activity. You can meet new friends or bring your own. I'd go myself if I could find someone to run the store. Due to the fuel shortage, we have reduced rates on every tour. Here's a pricelist." Xelda handed Vivian a pricelist.

"It's hard to make a living in this business without any fuel," she said. "I should have stayed in souvenirs." Xelda put out her long cigarette and lit another one.

Vivian looked at Xelda's pricelist. It had ten tours listed as follows (The Planet X symbol for xents is "x"):

1. Around the mountain 75x
2. 10-day cruise 150x
3. Scenic walk 40x
4. Xamster express 50x
5. Lake Lizard tour 60x
6. Cave exploration 35x
7. Week under the suns 80x
8. Town tour 100x
9. Fun on the rocks 60x
10. Creature comforts 40x

"Very interesting," said Vivian. "I wonder if you could tell me..." Vivian didn't have time to finish her sentence, however, because just then Xax and Xon came running through the door.

"There you are!" said Xax as he grabbed Vivian by the arm.

"Sorry for the disturbance," said Xon to Xelda, as he grabbed Vivian by the other arm.

"Stop!" said Vivian as she struggled to get free. "I want to go back to my spaceship."

"You can't leave now," said Xon. "You're the visitor from the sky who will help us. You must come with us to see Xaviar."

"I've got to move to a better neighborhood," said Xelda as she resumed reading her magazine.

Xax and Xon took Vivian outside to a parked propeller car. "This one has fuel in it," said Xax as he picked Vivian up and put her in the front seat. "Let's get out of here before the guards come."

Computers are often used to keep track of things like pricelists. We will write a program that accesses Xelda's tour pricelist.

An ordered list of numbers is called an *array*. To store an array in the computer's memory, you must first

dimension it so that the computer will know how many “elements” or numbers there are in the list. The DIM statement is used to give the maximum number of elements in the array, just as it was used to give the maximum number of characters in a string variable. For example, if there will be no more than 10 elements in an array, you can use a dimension of 10 (or any dimension larger than 10). An array name is like a name for a numeric variable. It must begin with a letter and can contain only capital letters and numbers. To specify a certain element in the array, you give the array name followed by the number of elements in parentheses. For example, if the tour pricelist array is denoted by P, the price of the sixth tour would be P(6).

The following program will store Xelda’s tour pricelist and enable you to find the price for one or more creatures to go on a particular tour.

```
10 DIM P(10)
20 P(1) = 75:P(2) = 150:P(3) = 40
30 P(4) = 50:P(5) = 60:P(6) = 35
40 P(7) = 80:P(8) = 100:P(9) = 60
50 P(10) = 40
55 PRINT
```

```
60 PRINT "WHICH TOUR # DO YOU WANT? "  
70 INPUT T  
80 IF T > 10 THEN GOTO 60  
90 IF T < 1 THEN GOTO 60  
100 PRINT  
110 PRINT "HOW MANY WILL BE GOING? "  
120 INPUT N  
130 PRINT  
140 PRINT "THE PRICE OF TOUR "; T;  
150 PRINT " IS "; P(T) ; " XENTS"  
160 PRINT  
170 PRINT "YOUR TOTAL COST WILL BE ";  
180 PRINT N*P(T) ; " XENTS"
```


You can easily change the values of individual elements in arrays. You can also expand an array into a *matrix* in which there is more than one number for each item in the list. For example, Xelda's pricelist could have a regular price and a discounted price for each tour. To have two prices for each tour, you would have a list of 10 *pairs* of prices. This information would be stored as a "10 by 2 matrix" and would be given the dimension DIM P(10,2). For example, the regular price and discount price of tour 7 would be P(7,1) and P(7,2) respectively.

FLIGHT TO THE HILLS

Xax tried to start the propeller car but the engine only sputtered and stopped. Just then, three lizard guards came running around the corner. "Hurry!" Xon urged Xax.

Xax tried to start the car again. After more sputtering, it started.

We will write a program in which the computer makes the sound of the propeller car. In ATARI® BASIC, sound



is created using the SOUND statement. Type this SOUND statement and press **RETURN**. To turn off the sound, press the **SYSTEM RESET** key.

```
SOUND 0, 25, 10, 8
```

The first number in the SOUND statement is the “voice.” There can be up to four voices at once: 0, 1, 2, or 3. To hear two voices type:

```
SOUND 0, 25, 10, 8
```

```
SOUND 1, 36, 10, 8
```

Press **RETURN** after each statement.

The second number in the SOUND statement determines the “pitch” or what note is played. This number can range from 0 to 255. Experiment with different sounds by varying the second number in the SOUND statement. For example, try these sounds:

```
SOUND 0, 10, 10, 8
```

```
SOUND 0, 80, 10, 8
```

```
SOUND 0, 130, 10, 8
```

The third number in the SOUND statement determines the tone quality. This number may be an even number between 0 and 14. Odd numbers turn off the sound. The number 10 is a pure tone. Some other num-

bers produce distorted tones and can be used for sound effects. To experiment with the possible sounds, use the following program:

```
10 PRINT "CHOOSE AN EVEN NUMBER"  
20 PRINT "FROM 0 TO 14"  
30 INPUT X  
40 FOR N = 0 TO 255  
50 SOUND 0, N, X, 10  
60 PRINT "SOUND 0", N, X, 10  
70 FOR L = 1 TO 50: NEXT L  
80 NEXT N
```

The fourth number in the SOUND statement determines the volume. This can be a number between 1 and 15, with 1 being the softest sound and 15 the loudest. 0 turns off the volume.

Now we will use the SOUND statement in a program that will demonstrate the sound of the propeller car. To clear the computer's memory, type NEW and press RETURN. The delay statement in line 60 is used to determine the length of time a sound is played. Enter and RUN the following program. The numbers printed on the screen are similar to the printout on the propeller car's dashboard computer.

```
30 FOR N = 0 TO 200
40 SOUND 0, 20, N, 8
50 PRINT N;
60 FOR Z = 1 TO 30: NEXT Z
70 NEXT N
```

You can vary the pitch of the propeller car with an INPUT statement. Make the following changes and RUN the revised program. Input different values for the pitch variable. (Remember, this number can range from 1 to 255.)

```
20 INPUT A
40 SOUND 0, A, N, 8
```

Next, you can change the “speed” of the engine by varying the length of delay in line 60. Make these changes and input different numbers for the pitch and the delay (Use a delay of 30 to start with).

```
10 INPUT B
60 FOR Z = 1 TO B: NEXT Z
```

The lizard guards ran across the street just as the propeller car lurched forward and around the corner. The car almost crashed into a listless crowd of lizard creatures,

narrowly missed a giant xamster, but managed to take Vivian, Xax, and Xon to the edge of town. Soon they were bouncing along a bumpy dirt road on their way into the mountains.

ATARI® BASIC has statements that let you create graphics on the screen. There are nine GRAPHICS “modes” numbered 0 through 8. You use these modes with the GRAPHICS statement followed by the number of the mode you want. When you don’t specify a GRAPHICS mode, you are automatically in GRAPHICS mode 0. This is called the “default” mode. GRAPHICS modes 1 through 8 have a split screen, in which there is a “window” for writing under the graphics screen. After you have entered a GRAPHICS mode and wish to return to the regular screen, you can either type GRAPHICS 0 and press **RETURN** or press the **SYSTEM RESET** key.

When you are in a GRAPHICS mode you can use a variety of foreground and background colors. Tables showing the colors available in the different GRAPHICS modes are in the back of the book.

To illustrate the route of the propeller car, we will use GRAPHICS mode 3. In this mode, the graphics screen is divided into 20 rows with 40 graphics locations in

each row. The 40 X 20 "points" on the screen are identified by a coordinate system with point 0,0 at the upper left-hand corner of the screen. To plot a point on the graphics screen, use the PLOT statement. For example, the statement

```
PLOT 18, 5
```

will plot a point on the graphics screen at the intersection of column #18 and row #5. To use the PLOT statement, you must first identify the GRAPHICS mode and state a color for the point to be plotted. To select a color, you use the COLOR statement. In GRAPHICS mode 3 there are three foreground colors available, numbered 0, 1, and 2. To experiment with the PLOT and COLOR statements, RUN the following program, substituting various values for colors and points to be plotted. Press **SYSTEM RESET** to return to the original screen.

```
10 GRAPHICS 3  
20 COLOR 1  
30 PLOT 18, 5
```

To plot the path of the propeller car, we will have the car flash on and off and "move" across the screen. This kind of motion is called *animation*. Cartoons are

made in much the same way. In animation, nothing actually moves. Instead, a series of pictures is drawn and erased very fast. This gives the appearance of movement. To erase a picture, you simply draw it using COLOR 0, which is the same as the background color. We will start the program by locating the propeller car at point 4,4. Type NEW and press RETURN. Then enter and RUN this program:

```
20 GRAPHICS 3
90 COLOR 1
100 PLOT 4, 4
```

Now we will have the car flash on and off by erasing and redrawing it at the same location. Make these additions and RUN the revised program. The delay statements cause the flashing effect.

```
80 FOR N = 2 TO 37
110 FOR W = 1 TO 20: NEXT W
130 COLOR 0
140 PLOT 4, 4
150 FOR W = 1 TO 20: NEXT W
160 NEXT N
```

Now we will add sound and have the propeller car move across the screen. Enter these lines and RUN the revised program.

```
100 PLOT N, 4
120 SOUND 0, 22, N, 8
140 PLOT N, 4
```

Now we will make the propeller car go across the screen and return. To do this, the computer must perform the above sequence of operations and then do it in reverse. A FOR/NEXT loop will have the computer count in reverse if you use the STEP variation of the FOR/NEXT statement. For example, the statement

```
FOR N = 50 TO 1 STEP -1
```

will cause the computer to start the loop with $N = 50$ and go backwards in "steps" of -1 units. The next value of N will be 49, then 48, and so on, until the 50th execution of the loop, $N = 1$. The statement

```
FOR N = 2 TO 50 STEP 2
```

will cause the computer to count forward by 2. The loop will start with $N = 2$, then let $N = 4$, then $N = 6$, then $N = 8$, and so on until $N = 50$. STEP 1 causes the loop to be executed in the usual way.

To make the propeller car go back and forth across the screen, we will need a FOR/NEXT loop going from 2 to 37 and then one going from 37 to 2 (using STEP -1). Rather than repeating the same sequence of instructions twice, we will use what is called a subroutine. A subroutine may be useful when you have a set of instructions that you want to access more than once in a program. To access a subroutine, you use a GOSUB statement, which directs you to the line number that begins the subroutine. At the end of the subroutine, you use a RETURN statement, which sends the computer back to the line after the GOSUB statement. Using GOTO statements will accomplish the same thing, but if the subroutine is accessed many times in a program, the GOSUB and RETURN statements are more convenient.

Add the following lines to the program. List the program so you can see the way the subroutine is accessed by the two GOSUB statements. Then RUN the program. The END statement in line 70 tells the computer to stop after it RETURNS from the second execution of the subroutine.

```
30 X = 2: Y = 37: Z = 1
40 GOSUB 80
50 X = 37: Y = 2: Z = -1
```

```
60 GOSUB 80
80 FOR N = X TO Y STEP Z
170 RETURN
75 END
```

Finally, we will make the propeller car travel back and forth and move down the screen. To do this, we will add instructions to the subroutine that cause the propeller car to go down a small distance each time it reaches the edge of the screen. We will add another FOR/NEXT loop so that the computer will repeat the sequence of going across and down, then back and down enough times to travel all the way down the screen. Make the following revisions and RUN the revised program.

```
22 FOR L = 1 TO 5
70 NEXT L
100 PLOT N, Q
140 PLOT N, Q
170 FOR Q = Q TO Q + 1
180 PLOT N - 1, Q
190 NEXT Q
200 RETURN
```

Riding along the dirt road leading into the hills, Vivian noticed changes in the landscape. There were piles of large, jagged rocks along the roadside that looked like they might tumble down at any moment. Tall, purple trees partially blocked the light coming from the sun. The climb grew steeper. Every so often, Vivian heard animal noises coming from behind bushes or rocks. The road got narrower until, finally, the rocks and trees prevented the propeller car from going any further. Xax stopped the car.

"Are you sure this is the right road?" Vivian asked Xax.

"I think so," he said. Xax made a high whistling sound. A few moments later, a thin lizard with slick hair and a mustache darted out from behind a pile of rocks.

"Xax! Xon!" said the thin lizard creature. "Welcome home!" The lizard creature then looked at Vivian uncertainly.

"Xete, I'm so glad to see you!" said Xon as he embraced the thin lizard creature. Xon motioned toward Vivian. "This is Vivian," said Xon. "She is the visitor from the skies who has come to save Planet X!"

"I hope you're right," said Xete suspiciously. "We must take her to Xaviar without delay!"

Before Vivian could reply, Xete took her hand and led her through a thicket of bushes with Xax and Xon close behind. Soon Vivian and her companions were walking along a narrow path through a thick forest of purple trees.



XAVIAR

The small path Vivian and the lizard creatures were on became even narrower. Vivian was becoming exhausted. Branches with thorns brushed across her face and she had to stoop to avoid overhanging branches. Purple plants left a sticky substance on her clothes. She shivered with cold. It was late in the day and the sun was getting lower in the sky.

They were high in the hills now. Vivian wanted to stop and rest. She walked around a turn and the path came to an end, blocked by thick bushes.

"We're here at last," said Xon. Pulling apart some branches, he stepped into the bushes and disappeared from sight. Xete followed Xon through the opening and Xax motioned for Vivian to follow Xete. With difficulty, she squeezed her way through the hole in the bushes.

On the other side of the bushes was a clearing. At the far side of the clearing a jagged mountain of rocks extended out of sight into a purple mist. A large campfire was in the middle of the clearing. Busy lizard creatures hurried back and forth from the clearing to the caves at the base of the mountain. As Vivian walked toward the campfire, the lizard creatures came to greet her companions.

When they reached the campfire, a large, handsome lizard with thick, wavy hair walked over and said, "Welcome back Xax and Xon, I see that you brought a visitor."

"Yes we have," said Xon. "Xaviar, this is Vivian, the visitor from the skies whom you promised would come."



"I'm so glad you're here," said Xavier. "I knew you would come because I wished that you would, and all my wishes come true. You are here to find fuel for us so that I can become the ruler of Planet X. By the way, what took you so long?"

Vivian told Xavier that the only reason she came to Planet X was because she received his help message, but that after giving the matter some thought, she really didn't think she could help him.

"Help message?" said Xavier. "I didn't send any help message ... But anyway, who cares? Now that you're here, you will find fuel for us." Xavier motioned with his hand. "See the beautiful landscape. See the Monday sun setting in the sky. We all want to live happily here. We need rulers like me who understand the needs of the citizens. We need fuel."

We will write a program that displays the Planet X landscape and the changing colors of the sky caused by the setting Monday sun. To do this, we will use the SETCOLOR command. In ATARI® graphics, this command controls certain aspects of the screen color.

The SETCOLOR command is followed by three numbers, as in "SETCOLOR 1, 4, 6" or "SETCOLOR 2, 7,



8.” The first number determines the color register in the computer that the command refers to. The color registers contain color information that the computer uses in the different GRAPHICS modes. The use of the color register depends on the GRAPHICS mode. In GRAPHICS mode 0, which we shall use, color register 2 determines the background color. The second number in the SETCOLOR command determines the color (hue) for the specified color register. This can be any number from 0 to 15. There is a table of colors in the back of the book. The third number determines the brightness (luminance) and can be any even number between 0 and 14. The higher the number, the brighter the color. We will use the SETCOLOR command to vary the background colors. Further use of this command is complicated, as the color registers have different uses in the different GRAPHICS modes.

We will create the Planet X landscape by using a GRAPHICS “control character” that looks like a tree. To use a GRAPHICS control character, you press the **CTRL** (Control) key on the left side of the keyboard and then, while the **CTRL** key is still down, press the character you want. (In Chapter 1, we used the **CTRL** key to activate



the buzzer.) The keys that have control characters are the letter keys along with “;”, “,”, and “.”. Try typing control characters. Notice that the control character for the “;” key looks like a tree. (Actually, it is the “spades” symbol used in cards.) We will use this symbol to create a landscape of trees. When you see CTRL ; in the program press ; while the **CTRL** key is pressed.

The first thing we will have the program do is print the landscape. Enter and RUN this program:

```
20 FOR I = 1 TO 911
30 PRINT "CTRL ; ";
40 NEXT I
190 GOTO 190
```

Next we will use the SETCOLOR command to illustrate the colors caused by the setting sun on Planet X. Add these lines and RUN the revised program.

```
50 FOR J = 0 TO 15
60 SETCOLOR 2, J, 4
70 FOR M = 1 TO 150: NEXT M
80 NEXT J
```

Finally, we will add sound to the program. Planet X is inhabited by a small insect called a “xrocket” that

makes a noise surprisingly like the chirp of a cricket. We will add a subroutine to the program that simulates the sound of a cricket. Instead of making the cricket chirp at regular times, we will access this subroutine at "random." We can do this with the RND(0) statement. We will pick a number from 0 to 3 at random and access the chirp subroutine only if the selected random number is 1. This means that there is a 1 in 4 chance of hearing the cricket chirp whenever line 52 is executed. Add the following lines and RUN the final version of the program.

```
100 FOR Y = 1 TO 2
110 FOR Z = 1 TO 10
120 FOR N = 5 TO 7 STEP 2
130 SOUND 0, N, 10, 10
140 NEXT N
150 NEXT Z
160 SOUND 0, 0, 1, 10
170 FOR W = 1 TO 99: NEXT W
180 NEXT Y
190 RETURN

52 R = INT(4*RND(0))
55 IF R = 1 THEN GOSUB 100
90 GOTO 50
```


As the Monday sun set in the Planet X sky, Vivian turned to Xaviar and said, "It will be dark soon. Tomorrow I'd like to go to my spaceship and be on my way home. I have to get back to Earth before school starts."

"It will be dark soon?" said Xaviar. "I'm not sure what you mean. It is only dark in caves and when we put blankets over our heads."

"Doesn't it get dark after the sun sets?" Vivian asked Xaviar.

"Don't be silly," said Xaviar, who then told Vivian about time on Planet X. "You see," he said, "there are two suns in the Planet X system, Monday and Yadnom. When the Monday sun sets, the Yadnom sun rises so that one sun is always in the sky. As one sun goes across the sky, the other sun goes under the planet to get ready for the next day. Thus, there are only two days on Planet X, Monday and Yadnom, named after the suns. Today is Monday and tomorrow it will be Yadnom. After that it will be Monday again."

"We have two clocks," continued Xaviar, "one for each sun. The unit of time is the XSEC. There are 100 XSECs in a day. Each day goes from 100 to 200 o'clock. When a sun rises, the clock for that sun starts at 100 o'clock. When the sun sets it is 200 o'clock. When the sun has set, time for that sun goes backwards so that the next day will begin as usual at 100 o'clock." Xaviar looked at the sky and pulled a small watch from his pocket. "It is now 199 Monday time and 101 Yadnom time. In another XSEC,




the Monday sun will set and the Yadnom sun will appear.”

Moments later, the Monday sun began to disappear behind the horizon. As this happened another sun appeared on the other side of the sky. “There it is,” said Xaviar, “the Yadnom sun.” Xaviar looked at his watch. “It is now 200 Monday time and 100 Yadnom time. Good morning. Good evening. It is Yadnom morning. It is also Monday evening. Monday time goes backwards now and Yadnom time goes forward. It is now 199 Monday time and 101 Yadnom time. Soon it will be time for Monday dinner, which is also Yadnom breakfast. I hope you will join us.”

We will write a program that simulates a Planet X digital clock. In order to quickly demonstrate the forward-backward feature of Planet X time, an XSEC in our computer clock can't be quite as long as a real XSEC. By experimenting with delay statements, you can create an actual Planet X digital clock or even an Earth clock.

In writing this program, we will use READ and DATA statements. These statements allow you to store numerical or written information in a program. Whenever a READ statement is encountered, the computer will read

as many times as are specified from a DATA statement located at the end of the program (or anywhere else in the program). The number of DATA items requested by the READ statement can't be more than the number of items listed in the DATA statement unless a RESTORE statement is also used. The RESTORE statement instructs the computer to go back to the beginning of the DATA list for more DATA to read. In our clock program, the computer will read through the list of data twice so we will need to use a RESTORE statement.



To practice using the READ, DATA, and RESTORE statements, enter and RUN the following program. The program instructs the computer to read pairs of numbers and print them. The RESTORE statement causes the program to go back through the list of data when it is finished the first time.

```
10 FOR K = 1 TO 2
20 FOR I = 1 TO 2
30 READ A, B
40 PRINT A; B
50 NEXT I
60 RESTORE
70 NEXT K
80 DATA 6, 7, 8, 9
```

We will use GRAPHICS mode 2 for this program, since in this mode you can type oversized letters and numbers. This will make our clock easy to read. In this GRAPHICS mode, the computer leaves a “window” for ordinary text at the bottom of the screen. In order to print letters or numbers on the graphics screen, we must type “PRINT #6” instead of PRINT. The “#6” tells the computer to print on the graphics screen instead of in the text window. Finally, we shall eliminate the text window from the screen by adding 16 to the GRAPHICS

mode. We will use GRAPHICS 2 + 16 which is GRAPHICS 2 without the text window.

When we want to print letters or numbers in a specific place on the graphics screen, we must first position the cursor in the appropriate place. We do this with a POSITION statement. For example, if we want to print something at location 4,7 we would have the statement POSITION 4,7 precede the PRINT statement. This is like the PLOT statement.

In the clock program that follows, the underline symbols () specify how many spaces to leave in order to have a neat display. Don't type the underline symbol. Just leave a space. Finally, to leave the program and return to the regular screen, press the **SYSTEM RESET** key. Type NEW to erase any program in the computer's memory. Then enter and RUN the following program, which will create a Planet X digital clock.

```
10 GRAPHICS 2 + 16
20 COLOR 1
30 POSITION 0, 0
40 PRINT #6; "PLANET X TIME CLOCK"
50 POSITION 0, 4
60 PRINT #6; "  _ _ MONDAY _ _ _ _ YADNOM"
```

```
70 FOR Y = 1 TO 2
80 READ A, B, C
90 FOR N = A TO B STEP C
100 POSITION 0, 8
110 PRINT #6; " ___ "; N; " _____ "; 300 - N
120 PRINT #6; " __XSECS_____XSECS"
130 SOUND 0, N - 100, 12, 8
140 FOR Z = 1 TO 99: NEXT Z
150 NEXT N
160 NEXT Y
170 DATA 100, 200, 1, 200, 100, -1
180 RESTORE: GOTO 70
```

"So you see," Xaviar continued, pointing toward the Yadnom sun, "there's always a sun in the Planet X sky. On very rare occasions, the two suns cross in the sky at once. It is said that strange things happen then."

Just then, a lizard creature announced that Monday dinner-Yadnom breakfast was ready. Vivian sat down to eat with Xaviar and the other creatures, wondering why the lizards had time go backwards. The A.M. and P.M. system worked perfectly well on Earth, she thought. She bit into a small green fruit that tasted like a sour lime and realized that she was very tired and needed to get some sleep.

THE TROLLEY

Vivian finished her Monday dinner-Yadnom breakfast and thanked Xaviar for his hospitality.

"I'm glad you enjoyed your meal," said Xaviar. "There are advantages to living in the mountains, but it's very isolated. We seldom have visitors, especially from outer space."

"I'm tired," said Vivian with a yawn. "I'd like to take a nap. Tomorrow I've got a long way to travel, about 100,000,000 miles."

"Xete will show you an empty cave," said Xaviar.

Vivian thanked Xaviar and walked with Xete to the base of the mountain, where a network of caves provided homes for Xaviar and his followers. Xete showed the tired girl a small, cozy cave containing a bed made of purple moss. In a few XSECs, Vivian was fast asleep.

When Vivian woke up a sun shone brightly in the sky. Of course, one of the Planet X suns always shone brightly in the sky.

Outside the cave, Vivian found Xaviar talking to Xon and another lizard creature. "Good afternoon-good night," said Xaviar with a toothy lizard smile. "I hope you had a good sleep."

"I slept quite well, thank you," Vivian said. "But I have to be going soon. I was wondering if someone could show me how to get back to my spaceship."

"I'm afraid that's impossible," said Xaviar. "You see, I wished that you would come to Planet X and find fuel and, as you know, all my wishes come true."



"Why don't you just wish that you would find some fuel by yourself?" said Vivian. "I really have to get going."

"Don't get upset," said Xaviar. "I can't just wish anything. It has to be the proper type of wish. Anyway, I'll see that you get back to your spaceship, but first you must find fuel." Xaviar motioned to a husky lizard creature who walked over to Vivian and took her by the arm.

"Xaviar is our leader," said the husky lizard creature. "We do as he wishes and his wishes always come true."

Vivian knew that she was in a difficult situation. She couldn't get back to her spaceship without Xaviar's help and he wouldn't do this unless she helped him find fuel. She was upset that she still didn't know who sent the help message. She thought about her predicament. Then she had an idea.

"Well, Xaviar," she said. "I've decided that I'll help you find fuel."

"Good!" exclaimed Xaviar. "Great!" exclaimed all the lizard creatures standing near Xaviar.

"But, in order to help you find fuel," Vivian continued, "I must use my spaceship's laser scanner. I'll have to fly low over the countryside and scan for underground fuel deposits."

"Very well," said Xaviar, "but in order to make sure that you don't just leave, I'll go with you in your spaceship. Besides, I've always wished that I could ride in a spaceship. I may as well make that wish come true."

After conferring with the other lizard creatures, Xaviar turned to Vivian and said, "Xax, Xon, Xete, and Xerxes will go with us to your spaceship. Only you and I will ride in the spaceship. There used to be a trolley system that ran up and down the mountain. On the other side of the clearing is an abandoned trolley. We have a small amount of fuel saved for special occasions. We'll ride the trolley down the mountain. We must leave now to reach your spaceship by tonight-tomorrow morning. Follow me." With Xax, Xon, Xete, Xerxes, and Vivian running to keep up with him, Xaviar walked quickly across the clearing to a grove of tall trees.

The abandoned trolley was on the other side of the grove. It was sitting on a rusty track that went out of sight into the bushes.

"Here it is," said Xaviar. He walked into the bushes and emerged a few moments later carrying two large cans filled with fuel. "It's down hill most of the way," said Xaviar as he poured fuel into the trolley's fuel tank.

"All aboard," said the lizard leader when he was finished, ringing a rusty bell. Vivian and the rest of the lizard creatures climbed aboard the trolley. After a few unsuccessful attempts, Xax started the trolley's engine and with a sudden lurch forward, the trolley slowly chugged off into the bushes.

The trolley bounced through the forest along the rusty track, showering its riders with leaves and broken branches. The lizards cheered excitedly while Vivian crouched behind a seat

to avoid the flying foliage. They sped precariously down the mountain. The trolley came out of the woods and across a large field. At the other side of the field Vivian had a spectacular view. Below them were hills and jagged rock formations. Farther down, Vivian could see a large lake and below that she could see the city. Vivian would have some interesting stories to tell her friends on Earth, providing that she was able to get away from this strange planet.

Traveling down a hill, Vivian watched the beautiful landscape pass by. Tall trees adorned the hills. The lake grew closer and it seemed to Vivian that if the trolley continued on its present course it would go straight into the water.

"How are we going to cross the lake?" Vivian asked Xaviar. "Is there a boat somewhere?"

"Don't worry," said Xaviar. "There's a bridge around the next bend."

Sure enough, as the trolley careened around the next bend, the tracks led over the lake along a narrow bridge. The bridge hadn't been traveled on in quite some time and, as the trolley carried them over the lake, the bridge started creaking and swaying. Vivian nervously looked at the water below. She glanced at the lizard creatures and saw that they were also looking at the water. She looked straight ahead and was horrified to see that in the middle of the lake the bridge had caved in.

"Stop the trolley!" Vivian shouted. "The bridge has collapsed!"





We will write a program that illustrates the predicament of the passengers on the Planet X trolley. We will use GRAPHICS mode 7, which allows us to draw smaller pictures than lower modes.

First we will draw the track. To do this, we use the DRAWTO statement. The DRAWTO statement instructs the computer to draw a straight line from where the cursor is currently positioned to any specific location on the screen. We must specify a color. Then we position the cursor with a PLOT statement. Then we use the DRAWTO statement. Enter and RUN the following program. Press **SYSTEM RESET** to return to the regular screen.

```
10 GRAPHICS 7
20 COLOR 2
30 PLOT 0, 20
40 DRAWTO 128, 20
```

Next we will draw the trolley. To do this we will use a different color than the track. We first use the DRAWTO statement to draw the top and bottom lines of the trolley. Then we use the PLOT statement to draw a point at every other space to define the windows.

Add the following lines and RUN the program.

```
60 COLOR 1
70 PLOT 10, 19: DRAWTO 20, 19
80 PLOT 10, 17: DRAWTO 20, 17
90 FOR X = 0 TO 10 STEP 2
100 PLOT 10 + X, 18
110 NEXT X
```

Now we are ready to animate the trolley by drawing it, erasing it, redrawing it, etc., to make it appear to move along the track. We will also add sound to the program and make it go to where the bridge is washed out. Make the following changes and additions and RUN the revised program.

```
50 FOR N = 0 TO 129
60 COLOR 1
70 PLOT N, 19: DRAWTO N + 10, 19
80 PLOT N, 17: DRAWTO N + 10, 17
90 FOR X = 0 TO 10 STEP 2
100 PLOT N + X, 18
110 NEXT X
120 SOUND 0, 22, N, 8
130 IF N = 129 THEN GOTO 200
140 COLOR 0
150 PLOT N, 19
160 PLOT N, 17
```

```
170 FOR X = 0 TO 10 STEP 2
180 PLOT N + X, 18
190 NEXT X
200 NEXT N
```

Now we will see what happens to the trolley. Add these lines and RUN the finished program.

```
300 FOR W = 1 TO 50: NEXT W
310 FOR A = 19 TO 80
320 FOR C = 1 TO 0 STEP -1
330 COLOR C
340 PLOT 129, A: DRAWTO 139, A
350 PLOT 129, A-2: DRAWTO 139, A-2
360 FOR X = 0 TO 10 STEP 2
370 PLOT 129 + X, A - 1
380 NEXT X
390 SOUND 0, A, 10, 8
400 NEXT C
410 NEXT A
420 FOR V = 1 TO 100
430 SOUND 0, V, 0, 10 - (V/10)
440 NEXT V
450 END
```



The lizard creatures were busy watching the water and didn't notice that the bridge had collapsed until it was too late. Xax tried to apply the brakes but the trolley went off the end of the bridge and fell into the water below. Fortunately, the water was shallow. In fact, when a soaking wet and very upset Vivian climbed out of the trolley, she discovered that the water barely reached her knees.

A few XSECs later, Vivian and the lizard creatures were standing on the bank of the lake staring gloomily at the partially submerged trolley.

"I guess we should start walking," said Xaviar. "When we get to the edge of town we will get a propeller car and drive to the spaceship." After a brief rest, Vivian and the lizards began hiking toward the city.

THE ARMY LIZARDS

Wet, tired, worried, sore, and angry, Vivian trekked through the woods with her lizard companions. The Yadnom sun sank lower in the sky and Vivian knew that it soon would be Monday again. She was trying to think of a way to get into her spaceship and take off without Xavier when suddenly Xerxes turned and shouted, "Hide! Soldiers!"

The lizards dropped to the ground and crawled quickly into the bushes and under rocks. Vivian dropped to the ground also, but she could not slither the way lizard creatures could. Instead, she just lay still.

Soon Vivian heard a rustling sound in the bushes near her. She looked up and saw a group of lizard soldiers march by. She put her head down and held her breath until they passed. When she was sure the soldiers had gone, Vivian stood up and her companions came out from their hiding places.





"Be careful," whispered Xaviar as he motioned for the others to follow. Vivian crouched down as she followed Xaviar through the woods.

The girl and the lizards crept through the woods for what seemed like a very long time until they finally reached a clearing.

"On the other side of the clearing there is a trail that leads down to a farm," said Xaviar. "The farmer who lives there is an old friend of mine. We can use his propeller car. If he has no fuel we can hitch the car to a xamster. From there we'll go to the spaceship. Let's cross the clearing quickly."

Halfway across the clearing, Vivian heard a loud rumbling sound. Suddenly, a large tank appeared at one end of the clearing. The lizards in the tank saw Vivian and her companions and started to move toward them.

"Run!" shouted Xaviar above the roar of the tank. Vivian and the lizards started to run and the tank started shooting. With an enormous BOOM! a shell exploded near Vivian. Another explosion caused Xax to leap up in the air. A second tank came rumbling onto the field. Both tanks started shooting with Vivian and the lizards trapped between them.

We will write a program that shows the tanks firing at Vivian and the lizards, who are trapped between them. We will use GRAPHICS mode 3. The sound effect we

use will be similar to those in some video games. First we will draw the tanks. We will use the PLOT and DRAWTO statements for this. The tanks will appear to come in from opposite sides of the screen. The FOR/NEXT loop of lines 30 and 120 causes this effect. Now enter and RUN the following program.

```
10 GRAPHICS 3
20 COLOR 1
30 FOR T = 1 TO 7
40 SOUND 0, 200, 6, 10
50 FOR A = 0 TO 2
60 PLOT 1, 17+ A: DRAWTO T, 17+ A
70 IF A = 1 THEN DRAWTO T+ 2, 18
80 PLOT 38, 17+ A: DRAWTO 39-T, 17+ A
90 IF A = 1 THEN DRAWTO 37-T, 18
100 NEXT A
110 FOR W = 1 TO 99: NEXT W
120 NEXT T
```

Now we will have the tanks shoot toward the middle of the field where Vivian and the lizard creatures are located. We will use COLOR 2 to draw the bullets. A bullet is just the cursor plotted at a particular location. To have a bullet move, we plot it and then erase it by

plotting it at the same location using COLOR 0. Then we plot it again slightly further along in its trajectory, then erase it, then plot it again, and so on. The sequence of drawing and erasing is done so fast by the computer that what you see appears to be bullets moving. Line 150 makes the sound of gunfire. Now add the following lines and RUN the revised program.

```
130 FOR V = 3 TO 9
140 FOR N = 0 TO V
150 SOUND 0, 20, 0, 10-N
160 COLOR 2: PLOT 10 + N, 18: PLOT 29-N, 18
170 FOR W = 1 TO 50: NEXT W
180 COLOR 0: PLOT 10 + N, 18: PLOT 29-N, 18
190 NEXT N
200 COLOR 2: PLOT 10 + N, 19: PLOT 29-N, 19
210 FOR W = 1 TO 50: NEXT W
230 COLOR 0: PLOT 10 + N, 19: PLOT 29-N, 19
240 NEXT V
```

Finally, we should see what happened to the two tanks. Enter the following lines and RUN the finished program.

```
250 SOUND 0, 20, 0, 10
```



```
260 FOR W = 1 TO 50: NEXT W
270 PLOT 1, 17: DRAWTO 38, 17
280 PLOT 1, 18: DRAWTO 38, 18
290 FOR W = 1 TO 50: NEXT W
300 PLOT 1, 19: DRAWTO 38, 19
310 FOR N = 1 TO 9
320 SOUND 0, 20, 0, 9-N
330 FOR W = 1 TO 50: NEXT W
340 NEXT N
```

Suddenly, the ground gave way and both tanks sank out of sight into the mud. The field was wet from recent rains and couldn't support the weight of the tanks. Vivian and her companions cheered. "Move fast before more of them come!" shouted Xaviar as he and his followers scurried toward the woods.

Soon they came to another clearing. An old farmhouse was at the far end of the clearing. A propeller car was parked in front of the house. Three xamsters grazed in a field next to the house. A group of small chicken-like birds with purple and green feathers were squawking and hopping around.

As Vivian approached the house she saw a lizard sitting in a chair on the front porch. He was wearing a large hat and a piece of straw was hanging out of the corner of his mouth.



He reminded Vivian of her Uncle John, who was a farmer in California.

The lizard farmer saw the visitors and stood up, squinting at them. "Xaviar!" he said when he recognized Xaviar, "I haven't seen you in many a Yadnom. What brings you here? Isn't it dangerous for you to come down from the mountains? If the forces of the Leadership Council find you here they will put you in jail or do something even worse."

"Greetings, Xim," said Xaviar. "I'm here because my wish to become ruler of Planet X is finally going to come true. We just evaded a group of soldiers from the Leadership Council. Just as I wished, Vivian has come from outer space to help us find fuel. We need to go to her spaceship so she can use her equipment. May I borrow your car?"

"I reckon so," said Xim. "I don't have any fuel though. You'll have to hitch it to a xamster."

"Fine," said Xaviar, who was already walking toward one of the grazing xamsters.

Soon Xim's propeller car was hitched up to a xamster. "Whoa, boy," said Xim, as he calmed down the protesting beast.

"I'll return your car as soon as possible," said Xaviar.


"Take your time," said Xim. "I wasn't planning to go anywhere."

Xax, Vivian, and Xaviar squeezed into the front seat of the propeller car and the other lizards climbed onto the hood and trunk. "Giddyap!" shouted Xaviar as he snapped the reins.

The xamster reared back and suddenly lurched forward. Vivian fell back in her seat and Xete toppled off the side of the propeller car into the bushes. Xete quickly brushed himself off and climbed back on the car and the xamster pulled the car down the last hill to the city. "We'll go around the outskirts of town to get to the road that leads to the spaceship," said Xaviar.

Buildings came into sight. Vivian and the lizards entered the outskirts of the city.

We will write a program that shows the buildings of the city as Vivian saw them when she and her lizard companions drove down the hill. This program illustrates the computer's ability to quickly and accurately perform a set of instructions over and over. This is one of the reasons why computers are so useful. The computer will draw a number of buildings, accompanied by building music. By using FOR/NEXT loops, a subroutine, and READ-DATA statements, the computer will be able to go back and use the same instructions over and over until six identical buildings are constructed. By tying the PLOT statement in with the FOR/NEXT loops, each new building is constructed next to the previous one. The PLOT and DRAWTO statements are used to draw each



building. The build instructions are in lines 20 through 80. The subroutine beginning at line 150 and continuing to line 200 provides the sound. The sound at the end is provided by lines 300 and 310.

Clear the computer's memory then enter and RUN the complete program. List the program when you are through running it and go over the different parts of the program as we have just described them.

```
10 GRAPHICS 3
20 FOR V = 3 TO 33 STEP 6
30 FOR N = 10 TO 19
40 COLOR 1
50 PLOT V, N: DRAWTO V + 4, N: GOSUB 150
60 IF INT(N/2) = N/2 THEN GOTO 80
70 COLOR 0: PLOT V + 2, N: GOSUB 150
80 NEXT N
90 RESTORE: NEXT V
100 IF V = 39 THEN GOTO 300
150 READ A, B
160 SOUND 0, A, 10, 10: SOUND 1, B, 10, 10
170 FOR W = 1 TO 50: NEXT W
180 SOUND 0, A, 11, 10: SOUND 1, B, 11, 10
190 FOR W = 1 TO 50: NEXT W
200 RETURN
```

```
300 FOR W = 255 TO 10 STEP -1
310 SOUND 0, W, 10, 10
320 NEXT W
490 DATA 170, 0, 130, 0, 90, 85, 90, 85, 0, 0
500 DATA 0, 0, 170, 0, 150, 0, 140, 0, 45, 42
510 DATA 45, 42, 0, 0, 0, 0, 140, 0, 135, 0
```

A low rumbling sound seemed to be coming from the direction they were headed. Growing nearer, Vivian recognized the sounds of music and marching and singing. Around a corner, Vivian and the lizards met a large group of lizard creatures marching through the streets. They were protesting against the Leadership Council and demanding more fuel.

BACK TO THE SPACESHIP

The protesters marched through the streets of the city demanding more fuel and a new government. "MORE ORE! MORE ORE!" they chanted, referring to the purple metal ore used to make fuel for Planet X machines. Vivian and the lizards watched from their propeller car as the marchers came down a street and stopped. A group of lizards with instruments were playing a strange melody. The marchers stopped in front of a funny looking lizard creature who had bulging eyes and looked something like a chameleon. He stood on top of a parked propeller car and was giving a speech.

"Hear me, fellow citizens!" said the lizard with the bulging eyes. "We are here today to protest the policies of a government that can't even find fuel to run the cars of the people!"

"Hear! Hear!" shouted the protesters.

"We have a government that puts its citizens in jail for suggesting new ways to do things... A government that has so many rules that citizens have made careers trying to figure them out!"

"Hear! Hear!" shouted the protesters.

"You see," said Xaviar to Vivian, "they want a new leader ... Someone who will help Planet X make progress."

"We have a Leadership Council that won't listen to anyone who wants to improve our planet," continued the lizard with the bulging eyes.



"I'll listen!" said Xaviar to Vivian.

"A Leadership Council that can't agree on what to do about anything!" said the lizard with the bulging eyes.

"I know what to do about everything!" said Xaviar.

"A Leadership Council made up of fools, not lizards!" said the lizard with the bulging eyes.

"I'm a lizard, not a fool!" said the excited Xaviar.

"We need a leader, not a Leadership Council!" shouted the lizard with the bulging eyes.

"And I am that leader!" said Xaviar to Vivian.

"And I am that leader!" shouted the lizard with the bulging eyes to the crowd.

"What?!" exclaimed Xaviar. "He can't do that! I'm going to be the next leader of Planet X."

"Hear! Hear!" shouted the crowd.

"Shh," said Vivian to Xaviar. "If the wrong lizards see you, they'll come get you. Anyway, we have to go to my spaceship. Once we find fuel the citizens will choose you to be their leader."

"I will be your leader!" said the lizard with the bulging eyes to the crowd.

"Hear! Hear!" shouted the crowd.

"You're right," said Xaviar to Vivian. "Once I find fuel, they'll want me to be leader. Let's go! Giddyap!"

The xamster, who had fallen asleep during Bulging Eye's speech, opened his eyes, looked around, and started trotting down the road.

We will write a program that illustrates the protest march. The music is complicated, so we will build the music part of the program first. Type and RUN the following program:

```
10 V = 50
60 FOR X = 1 TO 8
120 READ A, B, C: GOSUB 500
200 NEXT X
220 RESTORE: GOTO 60
500 SOUND 0, A, 10, 10: SOUND 1, B, 10, 8
510 SOUND 2, C, 10, 8
520 FOR W = 1 TO V: NEXT W
530 RETURN
900 DATA 230, 0, 0, 0, 0, 0, 115, 92, 75
910 DATA 0, 0, 0, 254, 0, 0, 0, 0, 0
920 DATA 127, 100, 85, 0, 0, 0
```

If you are so inclined, study the data entries and you will see how the music works. (The 0 gives no sound.) Now type in the following lines:

```
20 FOR P = 0 TO 9 STEP 9
40 SOUND 3, 28 + P, 10, 15: V = 50: RESTORE
210 NEXT P
220 RESTORE: GOTO 20
```

Note the delay number $V = 50$. The following statements will vary the delay. Add them to the program.

```
60 FOR X = 1 TO 40
100 IF X > 8 THEN V = 10
130 IF X = 8 OR X = 16 THEN RESTORE
140 IF X = 24 OR X = 32 THEN RESTORE
```

This is the protest music. To add the protesters, we shall use the control character **CTRL A** for the marchers. In entering the program, when you see **CTRL A** press A while the CTRL key is also pressed. The PRINT statements of lines 90 and 190 cause the marchers to march across the lower part of the screen. The PRINT statements of lines 30 and 110 cause the slogan "MORE ORE" to be printed. The underline symbols indicate how many spaces to leave. Now enter the remaining lines:

```
10 GRAPHICS 2: COLOR 1
30 POSITION 6, 3: PRINT #6; "MORE"
70 POSITION INT(X/2.1), 9
90 PRINT #6; "CTRL A__ CTRL A__ CTRL A__
CTRL A__ CTRL A__ CTRL A__"
100 POSITION 6, 3: IF X > 8 THEN V = 10
110 IF X > 4 THEN PRINT #6; "_ORE"
```



```
150 POSITION INT (X/2.1) , 9
160 IF X <> 40 THEN GOTO 190
170 FOR Z = 1 TO 40: NEXT Z
180 GOTO 200
190 PRINT #6; " [20 spaces] "
```

You may notice that, although the tempo of the music is approximately doubled, V varies from 50 to 10, or a factor of 5. This is because of all the other processing the computer has to do in addition to the music. You have to trust your ears when you're building a music program and find the right values. For example, in line 170 we need a higher delay number because the computer isn't printing anything at that point.

Vivian and the lizards went around another corner and came upon a contingent of soldier lizards who had been sent by the Leadership Council to monitor the protest march.

"Look!" said one of the soldiers. "It's Xaviar and the girl who escaped from jail!"

"Get them!" said another.



The soldiers started running toward the propeller car. Xaviar yanked on the reins, making the xamster's head jerk sideways. The xamster turned and ran down a side street with the soldiers in pursuit.

"Faster!" shouted Vivian. "If they catch us they'll put us all in jail."

The xamster was able to run faster than the soldiers who were all on foot. A few blocks later, the soldiers were far behind the propeller car.


"We must hurry," said Xaviar. "Soon the army will come in their cars." He snapped the reins and the xamster went faster.

As she rode along the same bumpy dirt road that she had been on when she first came into the town with Xang, Vivian anxiously wondered how she was going to escape from Planet X. She was afraid that the soldiers were going to catch up with them. Even if they didn't, she didn't know how she would get away from Xaviar.

After a long, uncomfortable ride, Vivian and the lizards reached the field where the spaceship had landed. The xamster pulled the propeller car across the field. Two other propeller cars appeared on a nearby hill. One was pulled by a xamster; the other was powered by its own engine. Both were filled with lizards.

"It's the army of the Leadership Council!" shouted Xon.

"It's the Leadership Council themselves!" shouted Xax.



Vivian looked at the approaching propeller cars. One of them was filled with lizard soldiers. In the other were four lizard creatures wearing pointed hats and dressed in purple robes. This car was driven by Xang. Vivian guessed that this was the Leadership Council.


The propeller cars came closer. One of the lizard soldiers boomed in a loud voice, "YOU ARE UNDER ARREST! DO NOT MOVE!"

Vivian realized that if the lizard soldiers put her in jail, she might never get away. She pressed a button on her pocket computer and a ladder descended from her spaceship. She scurried up the ladder and looked down to see Xaviar following right behind her. Then the lizards raised their weapons.

"STOP OR WE'LL SHOOT!" said a lizard soldier.

"Don't shoot!" said Xaviar. "This girl has come from outer space. She is going to find fuel for us. Let us go and search for fuel."

One of the robed lizards slowly stood up. He looked older than the other lizards. He motioned for silence before he solemnly spoke. "Xaviar," said the old lizard in a hoarse voice, "when your father and I were young we used to hunt xrickets together. We hid under rocks in a stream that long ago dried up. We played together in a forest where a city now stands. Life was different then. Food was plentiful and so was fuel. Your father and I didn't have a care in the world. Now things have changed.



Our fuel supply has dwindled to practically nothing. Our soil is poor. This is a time of hunger and scarcity.”

“You were a rebellious youth,” the old lizard continued. “Your father never knew what to do with you. You were his pride and joy. ‘Xaviar is so smart,’ he used to say, ‘but he doesn’t like anybody to tell him what to do.’”

The old lizard paused to clear his throat. “Now look at yourself,” he continued. “Leader of the Rebel Forces. Xaviar, it’s time to change. Join us on the Leadership Council. You’re a bright boy, Xaviar, and the citizens like you. Join us now and we won’t throw any of your followers in jail.”

“But Vivian and I will find fuel,” said Xaviar.

“We’ll find our own fuel,” said the old lizard. “We don’t know where this girl is from or why she is here. Who knows what she is planning? Come and join us.”

While the old lizard was talking to Xaviar, Vivian was quietly climbing up the ladder.

“The girl has come here because I wished that she would,” said Xaviar. “The trouble with you is that you never want to do anything new. You won’t listen to new ideas or think about things in different ways. You’re tired and feeble, old lizard, and the citizens are angry. They want someone else to be their leader and that someone will be me. I’m going with Vivian to find fuel and when I come back I will be the new leader of Planet X.”

Vivian had just reached the door of her spaceship when Xaviar finished talking. She opened the door and tried to climb inside the spaceship before Xaviar got up the ladder, but Xaviar was too fast.

"STOP!" ordered the old lizard, but Xaviar had already climbed into the spaceship and closed the door.


"Please, Xaviar," said Vivian inside the spaceship, "go back and be on the Leadership Council. They need you to help them and I need to go home."

"Don't be ridiculous," said Xaviar. "I'll go back and you can go home, but first we must find fuel."

Vivian saw that there was no way she could get Xaviar to leave the spaceship without first looking for fuel. "Very well," she said. "I'll circle the planet and use my laser scanner. But then I must be on my way. School starts in a few weeks and I can't be late. Sit there and fasten your seatbelt."

Xaviar and Vivian sat down and fastened their seatbelts. Vivian turned on the engines. With a mighty roar, her spaceship rose into the sky. The lizards below watched in amazement.

We will write a program that shows Vivian's spaceship blasting off from Planet X. When this program is finished, it will be the longest program in the book. We will build it in stages. The first part of the program



shows the spaceship before take-off with its engines warming up. The shape of the spaceship is irregular, so the PLOT and DRAWTO statements are used in a complicated way. Enter and RUN the following program. Line 170 stops the program without returning to the original screen. In line 10, "GRAPHICS" is abbreviated with "GR."

```
10 GR. 7
20 FOR X = 0 TO 49
30 R = 1: S = 1: COLOR 1
40 IF X > 10 THEN R = 5: IF X > 10 THEN S = 9
50 SOUND 0, 49 - X, 0, 10
60 IF X > 43 THEN V = V + 1
70 FOR Y = 43 - X + V TO 49 - X
80 PLOT 112 - Y - X, Y: DRAWTO 28 + X + Y, Y
90 NEXT Y
100 FOR Y = 50 - X TO 75 - X STEP R
110 PLOT 63, Y: DRAWTO 77, Y
120 NEXT Y
130 FOR Y = 70 - X TO 79 - X STEP S
140 PLOT 58, Y: DRAWTO 63, Y - 3
150 PLOT 82, Y: DRAWTO 78, Y - 3
160 NEXT Y
170 GOTO 170
```

Now we will have the spaceship blast off. Enter these lines and RUN the program.

```
170 COLOR 0
180 PLOT 63, 78 - X: DRAWTO 77, 78 - X
190 PLOT 58, Y - S: DRAWTO 63, Y - S - 3
200 PLOT 82, Y - S: DRAWTO 78, Y - S - 3
210 NEXT X
220 FOR Z = 0 TO 18
230 COLOR 1
240 PLOT 58, 21 - Z: DRAWTO 63, 18 - Z
250 PLOT 82, 21 - Z: DRAWTO 78, 18 - Z
260 COLOR 0
270 PLOT 63, 28 - Z: DRAWTO 77, 28 - Z
280 PLOT 58, 30 - Z: DRAWTO 63, 27 - Z
290 PLOT 82, 30 - Z: DRAWTO 78, 27 - Z
300 NEXT Z
```

We now adapt what we have done to get the rest of the ship to vanish.

```
310 FOR T = 0 TO 3
320 COLOR 1
330 PLOT 58, 3 - T: DRAWTO 63 - T, 0
340 PLOT 82, 3 - T: DRAWTO 77 + T, 0
350 COLOR 0
360 PLOT 58, 12 - T: DRAWTO 63, 9 - T
```

```
370 PLOT 82, 12 - T: DRAWTO 78, 9 - T
380 PLOT 63, 9 - T: DRAWTO 77, 9 - T
390 NEXT T
```

There go the tops of the tail fins!

```
400 FOR V = 0 TO 5
410 PLOT 58, 8 - V: DRAWTO 63, 5 - V
420 PLOT 82, 8 - V: DRAWTO 77, 5 - V
430 PLOT 63, 5 - V: DRAWTO 77, 5 - V
440 NEXT V
```

Almost gone!

```
450 FOR M = 2 TO 0 STEP -1
460 PLOT 58, M: DRAWTO 82, M
470 NEXT M
```

Now for the sound of the ship traveling away.

```
480 FOR E = 10 TO 0 STEP -1
490 SOUND 0, 0, 0, E
500 FOR W = 1 TO 100: NEXT W
510 NEXT E
```

ANOTHER CALL FOR HELP

Vivian was glad to blast off into space, even if Xavier was with her. When the spaceship was high enough, she leveled off and started flying over Planet X. They flew across a valley and above a mountain range. Vivian turned on her laser scanner and sent a beam across the planet's surface. If there were underground deposits of fuel, the laser beam would cause a signal to appear on the computer screen.

"I never knew that Planet X looked like this!" exclaimed Xavier as he peered through the viewport.


Vivian was busy scanning the planet for fuel. The spaceship passed over a mountain range, a lake, a forest, and rolling hills. Each time a new scene came into view, Xavier gasped in amazement.

"A city! Look at the tops of the buildings," he said.

Vivian circled the city a few times so Xavier could see it clearly.

Vivian watched the Monday sun sinking lower in the sky. Or was it the Yadnom sun? In any event, she wished that Xavier would stop distracting her so she could concentrate on finding fuel.

We will write two programs. The first program is a short one. It shows the laser scanner display. We will use GRAPHICS mode 3 + 16 which is GRAPHICS mode 3



without the text window. Lines 20 and 30 draw the screen rectangle. Line 80 keeps the program from going back to GRAPHICS mode 0. Now enter and RUN the following program.


```
10 GRAPHICS 3 + 16: COLOR 1
20 PLOT 5, 4: DRAWTO 33, 4: DRAWTO 33, 10
30 DRAWTO 5, 10: DRAWTO 5, 5
40 COLOR 2
50 FOR N = 0 TO 39 STEP 3
60 PLOT N, 1: PLOT N, 13
70 NEXT N
80 GOTO 80
```

To see the text window, modify and RUN the program as follows:

```
10 GRAPHICS 3: COLOR 1
75 PRINT "THE SPACESHIP 'S VIEWSCREEN"
```

The next program will show the tops of buildings as seen by Xaviar from the viewport when the spaceship passed over them. Type NEW to erase the previous program, then enter and RUN the following program.

```
10 GR. 3: SOUND 0, 10, 0, 10
20 COLOR 1
```



```
30 FOR N = 4 TO 34 STEP 10
40 PLOT N, 19: DRAWTO N + 5, 19
50 NEXT N
60 FOR X = 0 TO 4
70 FOR N = 4 TO 34 STEP 10
80 COLOR 1: PLOT N - X, 19
90 COLOR 0: PLOT N + 5 - X, 19
100 NEXT N
110 NEXT X
120 FOR X = 0 TO 5
130 COLOR 0: PLOT 5 - X, 19
140 FOR N = 10 TO 30 STEP 10
150 COLOR 1: PLOT N - X, 19
160 COLOR 0: PLOT N + 5 - X, 19
170 NEXT N
180 COLOR 1: PLOT 39 - X, 19: NEXT X
190 GOTO 30
```

Scanning the countryside with her laser scanner, Vivian wondered if the citizens of Planet X would want Xavier to be their leader. She hoped that Xavier would let her go home when they returned from their mission.

Vivian was daydreaming and listening to Xavier describe the view when the scanner display lit up with a series of red

blips. Vivian focused the scanner on the source of the blips and discovered that an underground fuel deposit was located at the base of a nearby mountain next to a lake. Further scanning indicated that the deposit was a large one.

"Xaviar!" said Vivian, as she maneuvered the ship toward the mountain. "I've found fuel!"

"Great!" shouted Xaviar as he jumped out of his seat and ran over to look at the laser scanner. "Where?" he asked.

"There is a large underground ore deposit at the base of the mountain," said Vivian, pointing to the blips on her screen. "There may be other deposits nearby."

Xaviar shook Vivian's hand. "We did it!" he exclaimed. "We found fuel! We saved the ... "

Suddenly the spaceship started violently shaking back and forth.

"What's going on?" said Xaviar as he held onto the instrument panel to keep his balance.

"I don't know," said Vivian, who was frantically trying to get to the computer. "There is some kind of air turbulence. If it doesn't stop we could be blown into the side of the mountain."

"Try to land," shouted Xaviar, "before the spaceship gets blown apart!"

It seemed as though the spaceship would indeed get blown apart if the shaking didn't stop. Vivian had never experienced such violent turbulence, not even in meteor storms or space quakes, and did not know what to do. Holding on to

whatever she could, she made her way across the room to the computer controls. She was stunned by what she saw, for the control screens were flashing on and off in a series of random patterns.

"Something weird is happening!" shouted Vivian to Xavier above the din. "The computer has broken down... I've lost control of the ship. I can't get a reading on our altitude or how fast we are traveling... or anything!"

The spaceship shook violently and started to spin. Xavier saw huge rocks and uprooted trees fly by the viewport. The spaceship flipped upside down so the viewport was pointing toward the sky. "Look!" shouted Xavier. "Both suns are in the sky! It's the Monday-Yadnom intersect!"

From her place at the computer control panel, Vivian could see the two suns through the viewport. "What?" shouted Vivian, who was sure the spaceship would blow apart at any minute.

"The Monday-Yadnom intersect — it's when both suns cross the sky at the same time," shouted Xavier. "It hasn't happened in my lifetime. I've read about it in books. Strange things take place but I can't exactly remember what."

The spaceship rocked up and down like a car going along an unpaved road. Vivian tried to steer the spaceship but the controls didn't function.

"Look at my watch!" shouted Xavier. The numbers on Xavier's Monday-Yadnom digital watch were flashing on and off.

"Time is out of control!" shouted the lizard. "I remember now! In a Monday-Yadnom intersect time ceases to exist."

Xaviar could say no more, for the spaceship now started spinning around as if it were caught in an enormous tornado. As they spun, Vivian thought of her parents and her friends on Earth and of how she could have been safely at home babysitting for little Harold Bloomgarden, who lived down the street, instead of being caught with a strange lizard in a world where time made no sense.

The spaceship spun faster and Vivian found herself forced against the wall. She couldn't remember where she was or what she was doing. With an enormous explosion, everything disappeared in a blur of light and sound. Then there was silence.

Enter and RUN the following program, which illustrates the enormous explosion on Planet X.

```
10 GR. 7 + 16
20 FOR N = 1 TO 16
30 COLOR N
40 FOR X = 0 TO 88 STEP 4
42 SOUND 0, 10 + 2*X, 0, 15 - X/6
43 SOUND 1, 0, 0, 10
50 PLOT 60 + 2*N, 45: DRAWTO 0, X
60 PLOT 60 + 2*N, 45: DRAWTO 2*X/1.2, 88
```



```
70 PLOT 60 + 2*N, 45: DRAWTO 150, X
80 PLOT 60 + 2*N, 45: DRAWTO 2*X/1.2, 0
90 NEXT X
100 NEXT N
110 GOTO 20
```

Later, Vivian wasn't sure whether she had been dreaming or not. She remembered that the spaceship had stopped bouncing and shaking and that there was quiet, calm darkness. She felt that there was no hurry to do anything, as if time had stopped. It seemed to Vivian that she could sit happily forever in the purple darkness of her spaceship.

Slowly, light returned. Vivian looked around the spaceship and saw Xavier sitting on the floor near her. At first she thought he was asleep or even dead but then he looked at her.

Vivian checked the instrument panel and saw that everything seemed to be working normally again. Looking through the viewport, she saw that they were hovering above the lake near where the fuel deposit had been discovered.

"What happened?" said Vivian.

"I'm not sure," said Xavier. "We started spinning and there was an explosion. That's all I remember."

Vivian went to the control panel and turned on the forward thrust rockets. The spaceship moved forward. Everything was normal again.

"I hope you use your new fuel wisely," said Vivian as she and Xaviar headed back to the field where they had left the other lizards. "Planet X needs a leader who will respect the environment and be sensitive to the needs of the citizens."

"Don't worry," said Xaviar. "once we get the machines running again there won't be any problems. They'll love me."

"They won't love you if you run out of fuel," said Vivian, who had doubts about whether Xaviar would be a responsible leader.

The spaceship came back to the field where they had taken off. As Vivian landed the craft, Xaviar looked through the viewport. With a blast of the stabilizer rockets, the spaceship came to a rest.

Vivian opened the door of the spaceship and the rays from a Planet X sun greeted her. Although Vivian wasn't sure which one it was, there was now only one sun in the sky.

"Well Xaviar, it's been nice knowing you," said Vivian as she shook the lizard leader's hand.

"You have made all of my wishes come true," said Xaviar as he gave Vivian a polite lizard kiss. "I do hope you'll come down and say goodbye to everybody."

Xaviar stepped through the doorway and onto the ladder. "What happened?!" he shrieked. "Something's wrong!!!"

Vivian looked through the doorway at the scene below. The lizard creatures were still there but they looked different.

Instead of wearing their fashionable lizard clothes, the

lizards were wearing primitive garments that looked like caveman clothes. The soldiers were carrying clubs, not rifles. The propeller cars were gone. In their place were xamsters with blankets thrown over their backs. The landscape was different too. There were more trees than before and there was a lake in the middle of the field.

"What happened?" whispered Vivian to Xaviar, who was standing next to her in the doorway of the spaceship.

"I think," said Xaviar slowly, "that we went backwards in time. I remember reading a book on Planet X mythology about how once in a while time stops and then reverses itself. I think it is supposed to happen during the Monday-Yadnom intersect. The two suns appearing together causes this reversal of time. We weren't affected by it because we were in the spaceship. But everything on the ground... "

Then the old lizard who used to be the leader of the Leadership Council stepped forward. He was wearing a robe made of tree bark.

"You from sky," said the old lizard, with a series of grunts. "What you do here?"

"I'm Xaviar... I've returned with fuel," said Xaviar uncertainly, as he stepped out on the ladder.

"Xaviar? Fuel?" said the old lizard curiously. "What means fuel?"

"You know, fuel," said Xaviar with panic in his eyes. "Fuel that is used to make machines run and cars go."



"Machines? Cars? What means machines? What means cars?" said the old lizard, who then began to laugh. Soon all the lizards on the ground were laughing hysterically.

"MACHINES! CARS! MACHINES! CARS!" the lizards chanted. Then the old lizard spoke again.

"Brother from sky," said the old lizard, "come show us how to fly. Tell us about machines and cars and fuel. Lead us."

"LEAD US! LEAD US!" chanted the lizards, as the soldiers pounded the ground with their clubs.

"You must go to them now," said Vivian to Xavier. "All your wishes have come true," she said. "You have found fuel and they want you to be their leader."

"But what good is fuel?" said Xavier. "There are no machines any more. And I don't want to lead lizards who went back in time."

"You'll be all right. You can teach them about machines and cars and fuel. They'll look up to you," reassured Vivian.

"MACHINES! CARS! FUEL! LEAD US!" chanted the lizards.

"But... But... I want to go with you," said Xavier to Vivian.

"Here," said Vivian. "Take this to remember me by." She handed Xavier her pocket computer.

"Oh, thank you," said Xavier. "But please don't go," he said. "Stay with me on Planet X. We can be rulers together."

"You know I can't do that," said Vivian. "Go now and be brave." She squeezed Xaviar's lizard hand and quickly stepped back into the spaceship.

The girl and the lizard took one last look at each other. Vivian pressed a button and closed the door of the spaceship. She watched Xaviar descend the ladder to the ground, then retracted the ladder and started the engines. Soon she was blasting off.

Traveling away from Planet X, Vivian thought about Xaviar and the other lizard creatures and her adventures on this strange planet. She was about to put her spaceship on automatic control and take a long nap when she noticed a message appearing on her computer communications screen. She gasped in surprise when she saw "HELP!" appear over and over on the screen.

Vivian discovered that the message was coming from where her spaceship had been just before the explosion. She suddenly realized the source of the message — her own computer!

There was a signal device on Vivian's computer that automatically broadcast a HELP message if the spaceship controls malfunctioned. The message must have been sent when the spaceship was spinning out of control.

Vivian remembered the Monday-Yadnom intersect and how time had gone backwards. The original HELP message Vivian

received that caused her to go to Planet X was, in fact, the same HELP message that she was receiving now. Since they had gone back in time, she had received the message before any of this had ever happened.

Vivian was getting a headache. She didn't want to think about Xaviar or Planet X or HELP messages or going back in time. With a flick of a switch, she turned off her communications computer. Moments later she was fast asleep.



GLOSSARY OF PROGRAMMING TERMS

(For more information consult
the ATARI BASIC REFERENCE MANUAL)

ARITHMETIC

In ATARI BASIC, the arithmetic operations are represented as follows:

- + addition
- subtraction
- * multiplication
- / division

Parentheses can be used to determine the order in which the operations are performed. The computer does the operation in the innermost parentheses first and works its way out. These operations can be used in a variety of ways, such as in the following programs:

```
10 INPUT X
20 A = (6 + 4 * (X - 3)) / 5
30 PRINT A

10 PRINT "INPUT 2 NUMBERS"
20 INPUT X, Y
30 PRINT X; "+ "; Y; "= "; X + Y
40 PRINT
50 PRINT X; "- "; Y; "= "; X - Y
```

```
60 PRINT
70 PRINT X; "*" ; Y; "=" ; X * Y
80 PRINT X; "/" ; Y; "=" ; X / Y
90 PRINT
100 GOTO 10
```

ARRAYS

An array is an ordered list of numbers. To store an array in the computer's memory, you must first tell the computer how many numbers or "elements" there are in the list. This is done with a DIM statement. For example, DIM A(5) means that the array denoted by A has no more than 5 elements. An array name is the same as the name of a numeric variable. It must begin with a letter and can contain only capital letters and numbers. A particular element in an array is referred to by giving the array name followed by the number of the element in parentheses. For example, A(4) refers to the fourth element of the array named A. The following program will store a pricelist in an array format. Here is the pricelist:

Item #	Price
1	\$1.75
2	2.50
3	6.25
4	1.25
5	3.10

```
10 DIM A (5)
20 A (1) = 1.75: A (2) = 2.50
30 A (3) = 6.25: A (4) = 1.25
40 A (5) = 3.10
50 PRINT "WHICH PRICE DO YOU WANT? "
60 INPUT N
70 PRINT
80 PRINT "PRICE "; N; " IS $"; A (N)
```

COLOR

The COLOR statement determines the color that will be used in the PLOT and DRAWTO statements. The COLOR statement is followed by a non-negative integer. COLOR 0 is the background color and will cause the computer to plot an invisible point. You should experiment with the

COLOR statement to see the effects. Alternating colors with COLOR 0 causes an object to have a shimmering effect. If something is being moved across the screen, drawing the image and “erasing” it by redrawing it in COLOR 0 gives an animation effect. Here is an example:

```
20 GRAPHICS 3
90 COLOR 1
100 PLOT 4, 4
```

To make the point flash on and off, add these lines:

```
80 FOR N = 2 TO 37
110 FOR W = 1 TO 20: NEXT W
130 COLOR 0
140 PLOT 4, 4
150 FOR W = 1 TO 20: NEXT W
160 NEXT N
```

To make the point flash on and off and move across the screen, make these changes:

```
100 PLOT N, 4
140 PLOT N, 4
```

CONT

If you interrupt execution of a program with the BREAK key, you can resume execution of the program by typing CONT and pressing the RETURN key.

DIM

The DIM (dimension) statement is used to reserve enough space in the computer's memory to store a string variable. Every string variable in a program must be preceded by a DIM statement. The number in the DIM statement can exceed the length (in characters) of the corresponding variable, but not the number of characters in the DIM statement. Here is an example:

```
10 DIM A$ (50)
20 A$ = "WUFFLEGUMP"
30 PRINT A$
```

You may dimension more than one variable on a single line. For example:

```
10 DIM A$ (50) , B$ (60) , C$ (70)
```

DRAWTO

The DRAWTO statement causes the computer to draw a line from one location to another on the graphics screen in one of the GRAPHICS modes 3-8. The starting location is determined by a previous statement. The ending location is specified in the DRAWTO statement. Here is an example:

```
10 GRAPHICS 7: COLOR 1
20 PLOT 5, 4
30 DRAWTO 120, 65
```

END

The END statement causes the computer to end its execution of a program. Here is an example. (Also try running this program without the END statement.)

```
10 PRINT "I 'M TIRED"
20 PRINT "I WANT THIS PROGRAM TO END"
30 END
40 GOTO 10
```

EQUALITY

In ATARI BASIC, the equality relations are denoted as follows:

- = equals
- > greater than
- >= greater than or equal to
- < less than
- <= less than or equal to
- <> not equal to

Here is an example:

```
10 INPUT A, B
20 IF A > B THEN PRINT A; ">"; B
30 IF A <> B THEN PRINT A; "<>"; B
40 IF A = B THEN PRINT A; "="; B
50 IF A < B THEN PRINT A; "<"; B
```

In the case of string variables, these relations are the same as alphabetical order. Here is an example:

```
10 DIM A$(50), B$(50)
```

```
20 INPUT A$, B$
30 IF A$ > B$ THEN PRINT A$; ">"; B$
40 IF A$ <> B$ THEN PRINT A$; "<>"; B$
50 IF A$ = B$ THEN PRINT A$; "="; B$
60 IF A$ < B$ THEN PRINT A$; "<"; B$
```

FOR/NEXT

The FOR/NEXT statements tell the computer to repeat a sequence of instructions a specific number of times. Repeating a sequence of instructions is called a *loop*. For example, if you want the computer to print "SORRY FOR THE DELAY" 27 times, you could use the following program:

```
10 FOR N = 1 TO 27
20 PRINT N; " SORRY FOR THE DELAY. "
30 NEXT N
```

Each time the computer reaches the NEXT N statement the value of N is increased by 1 and the computer goes back to the FOR statement. When the value of N reaches the number specified in the FOR statement (in

this case 27), the computer is finished with the loop and proceeds to the statement after the NEXT statement.

If you want the computer to count in increments other than 1 in the FOR/NEXT loop you can follow FOR with a STEP instruction. For example, the instruction

```
FOR N = 4 TO 30 STEP 2
```

will have the computer start the loop with $N = 4$ and increase the value of N by 2 each time the loop is completed until $N = 30$. The following programs illustrate the use of the STEP instruction:

```
10 FOR K = 1 TO 9 STEP 2  
20 PRINT K  
30 NEXT K
```

```
10 FOR J = 3 TO 24 STEP 3  
20 PRINT J  
30 NEXT J
```

```
10 FOR L = 15 TO 0 STEP -1  
20 PRINT L  
30 NEXT L
```

By having no instructions between the FOR and NEXT statements you can cause a time delay. This is be-

cause the computer will run through the loop the specified number of times without executing any instructions. The following program illustrates this. You can experiment by changing the ending value of K from 100 to 200 to 500 and then leaving out line 20.

```
10 PRINT "HELLO";  
20 FOR K = 1 TO 100: NEXT K  
30 GOTO 10
```

GOSUB and RETURN

The GOSUB and RETURN statements instruct the computer to go to a certain set of instructions, called a *subroutine*, and then return to the program once this set of instructions is performed. The GOSUB instruction sends the computer to the subroutine and the RETURN statement causes the computer to return to the line following the GOSUB statement. Subroutines are useful if you want the computer to perform the same set of instructions at different times during the course of a program. Here is a program that illustrates the use of subroutines:

```
10 A = 6: B = 10: C = 9
```

```
20 GOSUB 100
30 A = 8: B = 9
40 GOSUB 100
50 A = 10: B = 2: C = 1
60 GOSUB 100
70 END
100 PRINT "A="; A; " B="; B; " C="; C
110 RETURN
```

GOTO

The GOTO statement instructs the computer to go to a specified line number in a program for its next instruction. Here is an example:

```
10 PRINT "I WILL NOT TALK IN CLASS"
20 FOR K = 1 TO 50: NEXT K
30 GOTO 10
```

GRAPHICS

The GRAPHICS command is used to select one of the nine GRAPHICS modes 0-8. GRAPHICS mode 0 is the

mode your computer is in during ordinary use. Modes 1-8 have a "split screen," that is, a graphics screen and a "text window" beneath it. Characters can be printed in modes 0, 1, and 2, while modes 3-8 are for graphics only. The screen formats are different for each mode, with mode 8 having the highest resolution (most locations on

Table 1
GRAPHICS SCREEN FORMATS

Mode	Columns	Rows (Split Screen)	Rows (Full Screen)
0	40	—	24
1	20	20	24
2	20	10	12
3	40	20	24
4	80	40	48
5	80	40	48
6	160	80	96
7	160	80	96
8	320	160	192

the screen) and mode 2 the lowest. The number of rows and columns in each screen is shown in Table 1.

In order to change GRAPHICS modes, the GRAPHICS command is used by typing GRAPHICS or the abbreviation GR. followed by the mode you want, like GRAPHICS 3 or GR. 5. To get back to mode 0 you can press the **SYSTEM RESET** key.

To eliminate the text window, you add 16 to the mode. For example, to use mode 3 without a text window you would use the statement GR. 3 + 16.

The GRAPHICS modes have different color capabilities. These are explained in the ATARI BASIC manual.

IF-THEN

The IF-THEN statement instructs the computer to perform an instruction if a certain condition is met. If the condition is not met the computer goes to the next line in the program. Here is an example:

```
10 INPUT X
20 IF X >= 0 THEN GOTO 40
30 PRINT "X IS NEGATIVE": GOTO 10
```

```
40 PRINT "X IS NOT NEGATIVE": GOTO 10
```

INPUT

The INPUT statement instructs the computer to receive information entered from the keyboard. The information will be stored in the computer either as a numeric or a string variable. The following programs illustrate the INPUT statement:

```
10 INPUT X
20 PRINT
30 PRINT "THE VALUE OF X EQUALS "; X

10 DIM A$(50)
20 INPUT A$
30 PRINT
40 PRINT "HERE IS A$: "; A$
```

INT(X)

INT(X) instructs the computer to drop the fractional part (if there is one) of the number X. Here are some examples of how this function works:

```
INT(3.14) = 3
INT(.7581) = 0
INT(2) = 2
INT(1758.9) = 1758
```

Here is a program that picks random integers between 1 and 6. This program simulates the roll of a die. First we use the RND(0) function to pick a random decimal fraction. Multiplying by 6 and adding 1 will give us a random number between 1 and 7. Using INT(X) to drop the fractional part will give us an integer between 1 and 6, the simulated roll of a die.

```
10 X = INT(6 * RND(0) + 1)
20 PRINT X;
30 FOR L = 1 TO 50: NEXT L
40 GOTO 10
```

LET

The LET statement instructs the computer to store information as a numeric or string variable. It is not necessary to use the word LET in the instruction. Here are two examples:

```
10 LET X = 20
20 PRINT X + 10

10 DIM A$(50)
20 A$ = "FROGS"
30 PRINT "I LIKE "; A$
```

LIST

Typing **LIST** and pressing the **RETURN** key causes the computer to list the program currently in its memory.

NEW

Typing **NEW** and pressing the **RETURN** key causes the computer to clear its memory of any program so you can start writing a new program.

PLOT

The **PLOT** statement is used in **GRAPHICS** modes 3 through 8 to display a point on the graphics screen. The

color of the plotted point is determined by a COLOR statement. The position on the graphics screen is determined by the X and Y coordinates available in that GRAPHICS mode. For example, PLOT 5,8 would plot a point at the intersection of column #5 and row #8 of the appropriate graphics screen. Here is an example:

```
10 GRAPHICS 5
20 COLOR 1
30 PLOT 15, 12
```

POSITION

The POSITION statement is used to place the graphics screen cursor in a specific location without actually printing anything. This statement is usually used in GRAPHICS modes 1 or 2 for printing characters on the graphics screen. Here is an example:

```
10 GRAPHICS 2
20 COLOR 1
30 POSITION 2, 4
40 PRINT #6; "MATH IS FUN"
```

PRINT

The PRINT statement instructs the computer to print a message on the screen. You must put the message to be printed in quotes as follows:

```
PRINT "WHAT'S IN A GAME?"
```

You can also print information stored in a variable, as in the following examples:

```
10 DIM A$(50) : A$ = "HELP"  
20 PRINT A$
```

```
10 A = 75  
20 PRINT A
```

You can do arithmetic with a PRINT statement without writing a program. Here are some examples:

```
PRINT 9 + 2  
PRINT 6 * 7  
PRINT 8 / 2
```

Commas and semicolons can be used to control the space after PRINT statements.

No punctuation after a PRINT statement causes the

computer to skip to the next line before printing anything else.

A comma after a PRINT statement causes the computer to skip to the next "TAB" position before printing anything else. ATARI tab settings are every 10 spaces across the line. If you are within 10 spaces of the end of a line, the computer will skip to the next line if the PRINT statement is followed by a comma.

A semicolon after a PRINT statement will cause the computer to start printing again in the space after it stopped printing. Here is an example:

```
10 PRINT
20 PRINT "OH NO! "
30 PRINT "NOT ANOTHER "; "DULL ";
40 PRINT "PROGRAM! "
50 PRINT "OH NO! ", "OH NO! "
```

PRINT #6

In GRAPHICS modes 1 and 2 there is a graphics screen and a text window. In these modes, you can print characters on the graphics screen. The PRINT statement will

print on the text window. To print on the graphics screen you use PRINT #6 followed by a semicolon. Here is an example:

```
10 GRAPHICS 2: COLOR 1
20 POSITION 1, 4
30 PRINT #6; "THIS SENTENCE IS"
40 POSITION 7, 5
50 PRINT #6; "FALSE"
```

READ and DATA

The READ statement instructs the computer to read from a list of information in the program. This information follows a DATA statement. The information can be read in either numeric or string variable form. The READ and DATA statements allow you to store information in a program that can be used in the course of the program. The READ statement can be used repeatedly in the program and will read information from the DATA statement or statements in the order in which it appears. Here is an example that illustrates the READ and DATA statements:

```
10 DIM A$ (50)
20 READ A, B
30 PRINT A, B
40 READ A$, B$
50 PRINT "PROGRAMMING IS "; A$; B$
60 DATA 8, 15, FUN
70 DATA (SOMETIMES)
```

RESTORE

The RESTORE statement sends the computer to the beginning of a DATA list to use the same data more than once. Here is an example:

```
10 DIM A$ (50) , B$ (50)
20 READ A$, B$
30 PRINT A$; " ARE FUN"
40 FOR V = 1 TO 100: NEXT V
50 PRINT B$; "ARE DULL"
60 FOR V = 1 TO 100: NEXT V
70 RESTORE
80 GOTO 20
100 DATA MACHINES, PEOPLE
```

RND(0)

RND(0) instructs the computer to select a decimal fraction at random. Here is an example that prints a sequence of random decimal fractions:

```
10 PRINT RND (0)
20 FOR V = 1 TO 100: NEXT V
30 GOTO 10
```

RUN

Typing RUN and pressing the **RETURN** key causes the computer to run the program currently in its memory.

SETCOLOR

The SETCOLOR command is used to control various aspects of the screen color, such as the background color. The command actually affects the *color registers*, which are places in the computer's memory that store color information. The SETCOLOR command is followed by three numbers. The first number determines the color

register to be changed and can be either 0, 1, 2, 3, or 4. The color registers have different uses in different GRAPHICS modes. For example, in GRAPHICS 0, the ordinary screen mode, color register 2 determines the background color, while in GRAPHICS 3, color register 2 determines the color of the graphics point. The second number in the SETCOLOR command determines the color for the specified color register. This can be any number between 0 and 15. The third number determines the brightness ("luminance") and can be any number between 0 and 14.

Experiment with SETCOLOR by using these programs. The first program goes through the spectrum of background colors. The second program goes through the luminances for each background color.

```
10 FOR J = 0 TO 15
20 SETCOLOR 2, J, 4
30 FOR M = 1 TO 100: NEXT M
40 NEXT J
```

```
10 FOR X = 0 TO 15
20 FOR J = 0 TO 14
30 SETCOLOR 2, X, J
```

```
40 FOR M = 1 TO 100: NEXT M
50 NEXT J
60 NEXT X
```

SOUND

The **SOUND** statement is used to generate sounds. This statement is used with four numbers: **SOUND W, X, Y, Z**. The first number is the “voice.” There can be up to four voices at once: 0, 1, 2, or 3. To hear two voices type:

```
SOUND 0, 25, 10, 8
SOUND 1, 36, 10, 8
```

Press **SYSTEM RESET** to stop the music. The second number determines the “pitch,” or what note is played. This number can range from 0 to 255. Try these sounds to vary the pitch:

```
SOUND 0, 10, 10, 8
SOUND 0, 80, 10, 8
SOUND 0, 130, 10, 8
```

The third number in the **SOUND** statement determines the quality and can be any even number between

0 and 14. Odd numbers turn off the sound. The number 10 is a pure tone. Some other numbers produce distorted tones and can be used for sound effects. Use the following program to experiment with tone qualities:

```
10 PRINT "CHOOSE AN EVEN NUMBER"  
20 PRINT "FROM 0 TO 14"  
30 INPUT X  
40 FOR N = 0 TO 255  
50 SOUND 0, N, X, 10  
60 PRINT "SOUND 0", N, X, 10  
70 FOR L= 1 TO 50: NEXT L  
80 NEXT N
```

The fourth number in the SOUND statement determines the volume. This can be a number between 1 and 15, with 1 being the softest sound and 15 the loudest. 0 turns the volume off.

VARIABLES

A variable is a device for storing information in the computer's memory. It is like a box with a label. This label or variable name must begin with a letter and must contain

only capital letters and numbers. A variable that will be used to store only numbers is called a *numeric variable*. A variable that will represent non-numeric information is called a *string variable* and must end with the dollar sign symbol `$`. String variables must always be dimensioned with a DIM statement so the computer will know the maximum amount of room to allow for storage of the desired information. Here are two examples:


```
10 DIM A$(50)
20 LET A$ = "FRANK"
30 PRINT "WHO IS "; A$; "?"

10 INPUT X, Y
20 PRINT
30 PRINT X; "+" ; Y; "=" ; X + Y
```

INDEX OF PROGRAMMING TERMS

(Page numbers indicate
introduction of the term)

Arithmetic Statements, 26
Arrays, 45
COLOR, 54
CONT, 40
DIM, 12, 45
DRAWTO, 80
END, 57
Equality Relations, 37
FOR/NEXT, 9
FOR/NEXT/STEP, 56
GOSUB-RETURN, 57
GOTO, 3
GRAPHICS, 53
GRAPHICS Control
 Characters, 63
IF-THEN, 35
INPUT, 13
INT(X), 34
LET, 35
LIST, 3
NEW, 3
PLOT, 54
POSITION, 71
PRINT, 2
PRINT#6, 70
READ-DATA, 68
RESTORE, 69
RND(0), 34
SETCOLOR, 62
SOUND, 23, 50
Variables, 12, 26



\$12.95

Michael Orkin Ed Bogas

SURVIVAL on PLANET X

with the ATARI® Home Computer

Follow the space adventures of Vivian, the teenage space explorer on your ATARI Home Computer. While traveling through space in her spaceship, Vivian answers a call for help from Planet X. When she reaches the planet's surface, she finds herself in many adventure-filled situations and **you** can help her out! As you read about Vivian's adventures on Planet X, you'll also learn how to program in BASIC. All the programs are about Vivian, the planet, and the lizard people that inhabit Planet X. By entering the programs on your ATARI Home Computer, you'll see: how Vivian escapes from jail after the Leadership Council puts her there; how she meets Xaviar, the Rebel Leader; the ride on the trolley to find her spaceship; how time is measured on Planet X; and how Vivian and her new-found friends escape from the army lizards. As designers of video games, the authors add arcade-like excitement to the basic ideas and techniques of programming. Not only will you learn about programming in BASIC but you'll also discover some of the tricks and techniques that are used in your favorite arcade games!

About the authors: Michael Orkin is a video game designer and professor of statistics in California. Ed Bogas has also done video game designing and is a composer and producer in the entertainment business.

RESTON PUBLISHING COMPANY, INC.
A Prentice-Hall Company
Reston, Virginia



0-8359-7412-X